
Introduction

Ce livre est un hymne �a l'Obscure, �a cet �etat de l'Esprit �elu par le fruit inconscient
de notre Ame. C'est un art con�cu pour faire naitre la passion. Aussi ces mots qui
ne sont qu'assemblages et imbrications de lettres vous paraitront sans signi�cation
aucune mais ils sont en fait la nature fondamentale qui r�egit l'univers de la r�ealit�e
et ces alin�eas qui sont noy�es dans ce brouillard de silhouettes noirs sont cette part
de rêve sans quoi le monde ne pourrait exister.
C'est le langage AWK, �etendue in�nie de verts paturages issus d'un savoir-faire
malthusien encore ignor�e. Mais comme c'est �etrange de se retrouver dans cet �etat
oubli�e dont il est possible de rêver les nuits de somnolences inconscientes.

La premi�ere partie de ce livre explique les principes essentiels de AWK de fa�con
simple et de telle sorte que l'utilisateur les assimile facilement.
La deuxi�eme partie de ce livre d�etaille chaque point du langage AWK. On pourra
s'y r�ef�erer apr�es avoir bien compris chacun des exemples de la partie 1.
La derni�ere partie de ce livre donne quelques applications r�eelles trait�ees par AWK.

Chapter 1

Les principes essentiels de AWK

1.1 Introduction et d�e�nitions

1.1.1 structure d'un �chier de donn�ees

Le langage de programmation AWK est tr�es e�cace dans la gestion de �chiers. En
g�en�eral, un �chier de donn�ees admet une structure coh�erente qui peut se d�e�nir de
la fa�con suivante :

champ 1 champ2 ::: champ n (ligne 1)
champ 1 champ2 ::: champ n (ligne 2)
::::: ::::: ::::: ::::: (ligne i)
champ 1 champ2 ::: champ n (ligne q)

Les exceptions d�ependent de la d�e�nition même du type de donn�ees et peuvent être
trait�ees lorsqu'elles sont parfaitement connues.

L'exemple de �chier de donn�ees qui sera utilis�e dans toute la partie 1 sera le
suivant :

Fichier �c.data :

Nom Nombre de Marque de Prix du

cigarettes cigarettes paquet

fumees par de 20

jour cigarettes

Serge 20 Toute 10.50

Fredo 15 Camel 10.50

Marc 10 Camel 10.50

Alain 10 Peter_Bleu 10.50

Agnes 20 Goldo_Leg 6.50

Isabelle 15 Royal_Ment 12.50

Pyr 0 Aucune 0

Thyll 10 Camel 10.50

LeSaint 0 Aucune 0

Quelques notions de champs et de s�eparateurs :

1 Dans notre exemple, chaque champ est s�epar�e par un ou plusieurs blancs. Le
s�eparateur peut être modi��e comme on le veut par la variable globale FS.
Pour l'instant, FS = " " .

2 AWK compte le nombre de champs d'une ligne courante; ce nombre est plac�e
dans la variable globale NF (dans l'exemple, NF = 4).

3 AWK compte le nombre de lignes du �chier d'entr�ee; ce nombre est plac�e dans la
variable globale NR (dans l'exemple, NR = 9).

Exemples : visualisation du �chier d'entr�ee

� imprime chaque ligne du �chier.

awk '{print}' fic.data ou awk '{print $0}' fic.data

Serge 20 Toute 10.50

Fredo 15 Camel 10.50

Marc 10 Camel 10.50

Alain 10 Peter_Bleu 10.50

Agnes 20 Goldo_Leg 6.50

Isabelle 15 Royal_Ment 12.50

Pyr 0 Aucune 0

Thyll 10 Camel 10.50

LeSaint 0 Aucune 0

� �ecrit chaque champ s�epar�e d'un blanc (virgule), ligne par ligne.

awk '{print NR,$1,$2,$3,$4}' fic.data

1 Serge 20 Toute 10.50

2 Fredo 15 Camel 10.50

3 Marc 10 Camel 10.50

4 Alain 10 Peter_Bleu 10.50

5 Agnes 20 Goldo_Leg 6.50

6 Isabelle 15 Royal_Ment 12.50

7 Pyr 0 Aucune 0

8 Thyll 10 Camel 10.50

9 LeSaint 0 Aucune 0

� �ecrit le nombre de champ de la ligne, le premier puis le dernier champ de chaque
ligne prise en entr�ee.

awk '{print NF,$1,$NF}' fic.data

4 Serge 10.50

4 Fredo 10.50

4 Marc 10.50

4 Alain 10.50

4 Agnes 6.50

4 Isabelle 12.50

4 Pyr 0

4 Thyll 10.50

4 LeSaint 0

1.1.2 ex�ecution d'un programme AWK

La commande "awk" est tout d'abord une commande UNIX et par cons�equent en
pr�esente tous les avantages (pipe, �ecriture de scripte shell avec des instructions awk).
Mais surtout, AWK est un v�eritable langage de programmation .
Ainsi, il y a 2 fa�cons g�en�erales d'ex�ecuter des instructions AWK:

1 pour des applications simples :

awk 'program' �chier1 ... �chier n
(liste des �chiers de donn�ees en entr�ee)

'program' est une liste d'instructions se pr�esentant sous la forme qui sera
d�e�nie dans la partie suivante : pattern f action 1; action 2; ... ; action n g.
La commande ex�ecute le programme, �chier par �chier, sur chaque ligne de
fa�con s�equentielle, ou s'il n'y a pas de �chier en entr�ee, prend le standard input
en tant que �chier d'entr�ee.

Exemple :

imprime les lignes o�u la chaine "Camel" apparait en troisi�eme champ, c'est-�a-
dire les personnes fumant exclusivement des Camels.

awk '$3 == "Camel" {print}' fic.data fic.data

Fredo 15 Camel 10.50

Marc 10 Camel 10.50

Thyll 10 Camel 10.50

Fredo 15 Camel 10.50 imprime une deuxieme fois

Marc 10 Camel 10.50

Thyll 10 Camel 10.50

2 pour des applications plus complexes :
On ouvre un �chier d'instructions awk et l'ex�ecution s'e�ectue par la com-
mande :

awk -f �chierprog liste optionnelle de �chiers de donn�ees

La commande ex�ecute s�equentiellement le programme se trouvant dans le
�chier programme sur chaque ligne des �chiers en entr�ee.

Exemple : Coh�erence des donn�ees

On met les instructions suivantes dans le �chier "�chierprog" :

NF != 4 {print $0, "le nombre de champ n'est pas egal a 4"}

$2 < 0 {print $0, "le nombre de cigarettes fumees doit etre positif ou nul"}

$4 < 0 {print $0, "le prix doit etre positif ou nul"}

$4 > 20 {print $0, "le prix des cigarettes est cher"}

On lance le programme par la commande :

awk -f �chierprog �c.data

1.1.3 structure d'un programme AWK

Nous avons vu pr�ec�edemment un exemple o�u AWK �etait employ�e comme une com-
mande UNIX et o�u des conditions et des actions �etaient combin�ees :

$2<0 {print ...}

Ces deux arguments d�e�nissent la structure d'un programme AWK.
Un programme AWK se pr�esente toujours de la fa�con suivante :

PATTERN f ACTION g
PATTERN f ACTION g

..................

Un pattern est une condition portant exclusivement sur une ligne du �chier d'entr�ee
ou sur le standard input.
Une action s'e�ectue lorsque la condition est valid�ee ou si elle est e�ectivement
ex�ecutable.

Exemples :

� imprime le nom des personnes qui fument.

awk ' $2>0 {print $1} ' fic.data

pattern action

Serge

Fredo

Marc

Alain

Agnes

Isabelle

Thyll

� la pr�esence de pattern ou d'action n'est pas obligatoire :
�ecrit toutes les lignes o�u le second champ est nul, c'est-�a-dire les lignes des personnes
ne fumant pas.

awk '$2 == 0' fic.data ou awk '{if ($2 == 0) print}' fic.data

Pyr 0 Aucune 0

LeSaint 0 Aucune 0

L'op�eration de base de AWK consiste �a �echantillonner la s�equence des lignes
d'entr�ee les unes apr�es les autres, recherchant les lignes qui sont "match�ees" par
chacun des patterns du programme. AWK travaille de fa�con s�equentielle d'o�u son
int�erêt pour la gestion de �chiers.

1.1.4 quelques exemples

Nous supposerons maintenant que les lignes d'instruction sont �ecrites dans le �chier
"�chierprog" et sont lanc�es par l'alias cr�ee pr�ec�edemment "a".
Nous donnons ici quelques exemples d'utilisation de format de sortie avec les deux
fonctions print et printf dont la derni�ere s'apparente exactement �a la fonction printf
du langage C.

� Imprime le nom des fumeurs et le nombre de cigarettes fum�ees en deux jours

$2>0 {print $1,$2*2}

Serge 40

Fredo 30

Marc 20

Alain 20

Agnes 40

Isabelle 30

Thyll 20

� printf (format, valeur 1, valeur 2, valeur 3,....)
printf cr�ee n'importe quel type de format en sortie.
Le format est une string contenant des %
Chaque % est associ�e �a une valeur : il y a autant de % que de valeur.

{printf ("argent depense apres 1 jour pour %s : F %.2f\n", $1,$2*$4/20)}

%s : $1 pour une string (nom)

%.2f : $2*$4/20 pour un nombre avec 2 chiffres apres la virgule

\n : retour chariot

argent depense apres 1 jour pour Serge : F 10.50

argent depense apres 1 jour pour Fredo : F 7.88

argent depense apres 1 jour pour Marc : F 5.25

argent depense apres 1 jour pour Alain : F 5.25

argent depense apres 1 jour pour Agnes : F 6.50

argent depense apres 1 jour pour Isabelle : F 9.38

argent depense apres 1 jour pour Pyr : F 0.00

argent depense apres 1 jour pour Thyll : F 5.25

argent depense apres 1 jour pour LeSaint : F 0.00

1.2 D�ebuter avec AWK : les fonctions et les vari-

ables usuelles

Cette partie contient quelques notions essentielles lorsqu'on doit programmer en
AWK.

1.2.1 aucune d�eclaration de variable et de type

L'un des principaux attraits du langage AWK est qu'�a la fois les variables utilis�ees
et leur type ne sont pas d�eclar�es et qu'une variable peut aussi bien avoir un type
correspondant �a une chaine de caract�ere ou �a un nombre ou même, elle peut corre-
spondre aux deux types. AWK convertit la valeur d'une chaine en un nombre, ou
vice-versa lors de l'ex�ecution du programme.

Puisqu'une variable peut avoir �a priori deux types, il faut faire attention quant �a
son utilisationmais cela prof�ere au langage une tr�es grande souplesse que ne peuvent
se permettre le langage C ou le Pascal.

Exemple :

$1 == "Fredo" {

Fredo=10

$2=$2+Fredo

string=$1 " a l'etat stone :"

print string, $2

}

Fredo a l'etat stone : 25

1.2.2 la s�election par les Patterns : le "Matching"

Les Patterns sont tr�es utilis�es pour s�electer les lignes d'un �chier qui nous sont utiles
: lorsqu'un pattern est v�eri��e, la ligne correspondante du �chier est s�elect�ee ou "
match�ee " et s'il y a une action qui lui succ�ede, alors l'action est ex�ecut�ee sur cette
ligne sinon la ligne est a�ch�ee en sortie.
Cette section donne plusieurs grandes cat�egories de patterns souvent employ�es :

1 : S�election par comparaison
Les symboles de comparaison usuels (<;<=;==; ! =; >=; >) peuvent être
utilis�es �a la fois sur des nombres et sur des chaines.

Exemple :

imprime le nom des personnes suivi de la chaine ": gros fumeur"

$2 >= 20 {print $1, ": Gros fumeur"}

Serge : Gros fumeur

Agnes : Gros fumeur

2 : S�election par calcul
On peut e�ectuer toutes les op�erations arithm�etiques les plus simples que ce
soit dans l'utilisation des patterns ou dans l'�ecriture d'une action.

Exemple :

pattern s�electionnant les lignes contenant les fumeurs qui d�epensent par jour
une somme sup�erieure ou �egale �a 10,50 F.

$2*$4/20 >= 10.50

Serge 20 Toute 10.50

3 : S�election par texte
On peut s�electer des lignes sp�eci�ant simplement un mot.

Exemple :

pattern s�electionnant toutes les lignes dont le premier champ est par ordre
alphab�etique avant la chaine "Fredo".

$1 < "Fredo"

Alain 10 Peter_Bleu 10.50

Agnes 20 Goldo_Leg 6.50

Mais aussi, il est possible d'e�ectuer des recherches sur n'importe quel type
de caract�ere en utilisant des expressions r�eguli�eres, similaires au expression
r�eguli�eres sous "ed" (nous d�etaillons cet aspect dans le chapitre 2).

Exemples :

� matche toutes les lignes o�u apparait la chaine "Camel" même sous la forme
"RallyeCamel" ou sous la forme "TETEDECamel".

/Camel/

Fredo 15 Camel 10.50

Marc 10 Camel 10.50

Thyll 10 Camel 10.50

� matche les lignes contenant le symbole y et les imprime.

/y/

Isabelle 15 Royal_Ment 12.50

Pyr 0 Aucune 0

Thyll 10 Camel 10.50

4 : Combinaison de patterns
Comme en C, les op�erateurs bool�eens AND, OR et NOT sont repr�esent�es par
&&, || et | .

&& ----> AND

|| ----> OR

| ----> NOT

Exemple :

pattern imprimant les lignes o�u le nombre de cigarettes fum�ees est sup�erieur
�a 10 ET o�u la marque est Camel.

$2 > 10 && $3 == "Camel" ou | ($2 <= 10 || $3 != "Camel")

Fredo 15 Camel 10.50

5 : BEGIN et END
Les patterns BEGIN et END jouent un rôle particulier et sont tr�es souvent
employ�es. Lorsque BEGIN est utilis�e (respectivement END), les actions qui lui
succ�edent sont e�ectu�ees avant (respectivement apr�es) que la premi�ere (respec-
tivement derni�ere) ligne du �chier d'entr�ee soit lue. Ainsi, on peut e�ectuer
des programmes AWK sans avoir n�ecessairement de �chier en entr�ee.

Exemple :

BEGIN {print; print "NOM NOMBRE MARQUE PRIX";print}

{print} # commentaires : action sur le fichier fic.data

END {print; print "Nombre de lignes lues :", NR

print "VISUALISATION TERMINEE"; print}

NOM NOMBRE MARQUE PRIX

Serge 20 Toute 10.50

Fredo 15 Camel 10.50

Marc 10 Camel 10.50

Alain 10 Peter_Bleu 10.50

Agnes 20 Goldo_Leg 6.50

Isabelle 15 Royal_Ment 12.50

Pyr 0 Aucune 0

Thyll 10 Camel 10.50

LeSaint 0 Aucune 0

Nombre de lignes lues : 9

VISUALISATION TERMINEE



1.2. D�EBUTERAVECAWK : LES FONCTIONSET LES VARIABLESUSUELLES19

1.2.3 la programmation par les actions : les tests et les

boucles

La syntaxe des actions est tout �a fait similaire �a celle du langage C mais la grande
di��erence r�eside dans le fait qu'il n'y a aucune d�eclaration de variable comme nous
l'avons dit pr�ec�edemment.
Nous donnons quelques exemples d'instructions usuelles.

1 : if-else
Les patterns et les actions de test ou de condition sont �a priori similaires mais
il faut bien voir qu'un pattern agit sur toutes les lignes d'un �chier en entr�ee
(r�eaction en chaine) alors qu'une action de test peut agir sur toute variable
tout en contrôlant l'ex�ecution du programme.

Exemple :

Moyenne sur l'argent depens�ee par fumeur et par jour.

$4 > 0 {n=n+1; argent=$2*$4/20 + argent}

END {

if (n>0)

{

printf (" Argent depense en moyenne par fumeur par jour :")

printf ("%5.2f pour %d fumeurs\n\n", argent/n, n)

}

else

print "pas de fumeur parmi les", NR, "personnes"

}

Argent depense en moyenne par fumeur par jour : 7.14 pour 7 fumeurs

2 : while
Exemple :

Sachant que le prix du paquet de cigarettes augmente tous les 6 mois de
10 %, donnons les prix apr�es 2 ann�ees.

BEGIN {printf("\n\t Marque\t6 mois\t12 mois\t18 mois\t24 mois\n")}

$4 != 0 {i=1; prix=$4;printf("\n\t%10s",$3)

while (i<=4)

{

prix=prix*1.1

printf("\t%4.1f",prix)

i=i+1

}

}

END {printf("\n\n")}

Marque 6 mois 12 mois 18 mois 24 mois

Toute 11.6 12.7 14.0 15.4

Camel 11.6 12.7 14.0 15.4

Camel 11.6 12.7 14.0 15.4

Peter_Bleu 11.6 12.7 14.0 15.4

Goldo_Leg 7.2 7.9 8.7 9.5

Royal_Ment 13.8 15.1 16.6 18.3

Camel 11.6 12.7 14.0 15.4

3 : for
Exemple :

imprime tous les fumeurs du �chier �c.data en cr�eant une tabulation (nt)

$3 != "Aucune" {for (i=1;i<=NF;i++) printf("\t%10s",$i)

printf("\n")

}

Serge 20 Toute 10.50

Fredo 15 Camel 10.50

Marc 10 Camel 10.50

Alain 10 Peter_Bleu 10.50

Agnes 20 Goldo_Leg 6.50

Isabelle 15 Royal_Ment 12.50

Thyll 10 Camel 10.50

1.2.4 les op�erations AWK sur les chaines de caract�eres

1 : la concat�enation de chaine
En AWK, on peut assembler des chaines de caract�eres sans que cela pose de
probl�emes en �ecrivant les chaines les unes �a la suite des autres.

Exemple :

assemble tous les noms des personnes du �chier �c.data dans la variable
"noms" puis imprime le r�esultat

{noms=noms $1 " "}

END {print noms}

Serge Fredo Marc Alain Agnes Isabelle Pyr Thyll LeSaint

2 : les fonctions AWK
Comme la plupart des langages, AWK permet l'utilisation de fonctions pr�e-
construites admettant des param�etres et retournant le r�esultat de leur ex�ecution.

� Les fonctions arithm�etiques sont donn�es dans le chapitre 2 et leur utili-
sation est tr�es simple:

Exemple :

�ecrit le logarithme n�ep�erien puis le logarithme d�ecimal du prix des paquets.

$4 != 0 {print "log du prix :",log($4),"log10 du prix :",log($4)/log(10)}

log du prix : 2.35138 log10 du prix : 1.02119

log du prix : 2.35138 log10 du prix : 1.02119

log du prix : 2.35138 log10 du prix : 1.02119

log du prix : 2.35138 log10 du prix : 1.02119

log du prix : 1.8718 log10 du prix : 0.812913

log du prix : 2.52573 log10 du prix : 1.09691

log du prix : 2.35138 log10 du prix : 1.02119

� Les fonctions sur les chaines sont sp�eci�ques au langage AWK puisque toutes
les variables sont d�e�nies comme des chaines de caract�eres.
Nous donnons ici deux des fonctions les plus employ�ees : split et gsub.

La fonction split(s,a) �eclate la chaine s, caract�ere par caract�ere suivant le
s�eparateur FS et place chaque �el�ement �eclat�e dans le tableau a (nous voyons
les tableaux plus bas). Elle retourne le nombre d'�el�ements du tableau a.

Exemple :

explosion de la ligne contenant la chaine Fr.

/Fr/ {max=split($0,tab)

for (i=1;i<=max;i++) print i, tab[i]

}

1 Fredo

2 15

3 Camel

4 10.50

La fonction gsub(r,s,t) substitue par s ce qui est match�e par l'expression
r�eguli�ere r, globalement dans la chaine t.

Exemple :

suppression du dernier caract�ere dans la chaine Fredo jusqu'�a sa �n d�e�nitive
et alors sortie forc�ee du programme par "exit".

$1 == "Fredo" {printf(" Mort lente de %s \n", $1)

chaine=$1

while ($1 != "")

{

printf("\t%s\n", $1)

gsub(/[a-zA-Z]$/,"", $1)

}

printf(" %s est mort mais il meurt avec $1\n", chaine)

exit

}

Mort lente de Fredo

Fredo

Fred

Fre

Fr

F

Fredo est mort mais il meurt avec $1

1.2.5 les tableaux - la notion de tableaux associatifs

AWK permet l'utilisation de tableaux pour stocker des chaines ou des nombres sans
qu'il y ait besoin de les d�eclarer.

Exemple :

�ecrit le �chier d'entr�ee dans l'ordre inverse.

{x[NR]=$0}

END {for (i=NR;i>0;i--) print x[i]}

LeSaint 0 Aucune 0

Thyll 10 Camel 10.50

Pyr 0 Aucune 0

Isabelle 15 Royal_Ment 12.50

Agnes 20 Goldo_Leg 6.50

Alain 10 Peter_Bleu 10.50

Marc 10 Camel 10.50

Fredo 15 Camel 10.50

Serge 20 Toute 10.50

La caract�eristique qui fait que les tableaux en AWK sont di��erents des tableaux
employ�es par les principaux langages est que les indices sont des chaines. Pour cette
raison, les tableaux sous AWK sont appel�es tableaux associatifs.

Exemple :

/Camel/ {Nb[Camel]+=$2}

/Peter_Bleu/ {Nb["Peter_Bleu"]+=$2}

END {print " Nombre de Camel fumees :", Nb[Camel]

print " Nombre de Peter_Bleu fumees :", Nb["Peter_Bleu"]}

Nombre de Camel fumees : 35

Nombre de Peter_Bleu fumees : 10

$3 ~ /Toute/ {

tab[6]="clo";tab[3]="x";tab[4]="eur "

tab[5]="de ";tab[2]="ta";tab[1]="pes"

printf("%s ",$1)

for (c in tab) printf("%s", tab[c])

print ""

}

Serge taxeur de clopes

Avec l'option "in", l'ordre d'ecriture est quelconque.

1.2.6 cr�eer une fonction par "function"

En plus des fonctions pr�e-d�e�nies, un programme AWK peut contenir ses propres
fonctions. De même qu'en C, une fonction se d�e�nit de la fa�con suivante :

function nom (liste de param�etres) f instructions g

La d�e�nition des fonctions est enti�erement s�epar�ee du corps du programme princi-
pal. Le corps de la focntion peut contenir l'instruction "return" qui retourne alors
un �etat ou une valeur vers l'instruction appelante.

Exemple :

Lorsque Fredo et Marc se rencontrent, il y a une consommation de cigarettes qui
croit par rapport �a la normale de fa�con g�eom�etrique (raison 2). On veut calculer au
bout de combien de temps chacune des deux personnes aura consomm�e 20 cigarettes
sachant qu'il y a 15 heures dans une journ�ee de fumeur :
On utilise pour cela une fonction calcul qui �evalue le temps en minute �a partir du
nombre de cigarettes fum�ees par jour.
On utilise aussi la fonction arithm�etique "int" qui prend la partie enti�ere d'une
valeur num�erique.

$1=="Fredo" {tab["fredo"]=calcul($2)}

$1=="Marc" {tab["marc"]=calcul($2)}

END {

printf ("Temps de rencontre apres 20 cigarettes fumees\n")

printf ("\t pour Fredo : %d heures", int(tab["fredo"]/60))

printf (" %d minutes\n", tab["fredo"]-60*int(tab["fredo"]/60))

printf ("\t pour Marc : %d heures", int(tab["marc"]/60))

printf (" %d minutes\n", tab["marc"]-60*int(tab["marc"]/60))

}

function calcul(nombre)

{

normal=nombre/15

terme=normal

x=0;i=0

while(x<20)

{

i++

xold=x

x=x+terme

terme=2*terme

}

cigaminute=(x-xold)/60

min=(20-xold)/cigaminute

temps=int((i-1)*60+min)

return(temps)

}

Temps de rencontre apres 20 cigarettes fumees

pour Fredo : 4 heures 18 minutes

pour Marc : 4 heures 56 minutes

1.2.7 les entr�ees : la fonction "getline "

Il y a plusieurs fa�cons de cr�eer des entr�ees �a un programme AWK. La fa�con la plus
simple est celle que nous utilisons depuis le d�ebut, c.a.d. placer des donn�ees dans
un �chier �c.data puis le lire s�equentiellement par les commandes :

- awk 'program' fic.data

- awk -f fichierprog fic.data

Une autre fa�con est d'utiliser les pipes :

Exemple : on utilise la commande grep pour s�electionner la ligne contenant "Fredo"
puis on travaille dessus.

grep 'Fredo' fic.data | awk '$2>0 {print "Fredo est un fumeur"}'

Fredo est un fumeur

La fonction " getline " lit l'enregistrement sur l'entr�ee et incr�emente NR. Elle re-
tourne 1 s'il n'y a pas eu d'erreur 0 si c'est la �n de �chier et -1 en cas d'erreur.

Exemples :

� lit la ligne point�ee du �chier "�chier" :

getline < "fichier"

� place dans la variable x la ligne point�ee du �chier "�chier" :

getline x < "fichier"

� tous les exemples pr�ec�edents peuvent être �ecrit avec la fonction getline en pla�cant
le corps du programme dans BEGIN :
Mort lente de Fredo.

BEGIN {

while (getline<"fic.data">0 && $1=="Fredo")

{

printf(" Mort lente de %s \n", $1)

chaine=$1

while ($1 != "")

{

printf("\t%s\n", $1)

gsub(/[a-zA-Z]$/,"", $1)

}

printf(" %s est mort mais il meurt avec $1\n", chaine)

exit

}

}

Mort lente de Fredo

Fredo

Fred

Fre

Fr

F

Fredo est mort mais il meurt avec $1

Le programme se lance alors par la commande :

awk -f fichierprog

1.2.8 les sorties et les int�eractions avec d'autres programmes

: la fonction "system"

Nous avons vu avec "print" et "printf" des exemples d'a�chage en sortie sur le
standard output mais il est possible aussi de diriger la sortie �a l'int�erieur d'un pro-
gramme awk vers un �chier.

Exemples :

le premier exemple imprime les noms des fumeurs dans un �chier nomm�e "nom".
le second exemple imprime �a la suite du �chier nom ainsi cr�e�e la marque de cigarette
fum�ee.

awk '$2>0 {print $1 > "nom"}' fic.data

awk '$2>0 {print $3 >> "nom"}' fic.data

fichier "nom" resultant

Serge

Fredo

Marc

Alain

Agnes

Isabelle

Thyll

Toute

Camel

Camel

Peter_Bleu

Goldo_Leg

Royal_Ment

Camel



28 CHAPTER 1. LES PRINCIPES ESSENTIELS DE AWK

Les int�eractions avec d'autres programmes se font de fa�con di��erente :

1 : les pipes en sortie et en entr�ee

print | commande

commande | getline

Exemples :
tri suivant le nom des fumeurs ; on utilise l'option "in" dans for ce qui nous
permet d'obtenir tous les �el�ements du tableau constitu�e.

$2>0 {Marque[$1]=$3}

END {

for (c in Marque)

{

printf("%8s \t%10s\n", c, Marque[c]) | "sort -b +0"

}

}

Agnes Goldo_Leg

Alain Peter_Bleu

Fredo Camel

Isabelle Royal_Ment

Marc Camel

Serge Toute

Thyll Camel

d�ecompte du nombre de personnes logg�ees sur le syst�eme Arthur �a un instant
donn�e.

BEGIN {

while ("who" | getline x) {n++;print x}

print "personnes loggees sur Arthur :", n

}

fredo ttyq1 Jun 12 15:28

bonhair ttyq2 Jun 12 17:21

gillou ttyq3 Jun 12 17:35

gemo ttyq4 Jun 12 16:01

jackie ttyq6 Jun 12 17:35

renz ttyq8 Jun 12 16:47

leptose ttyq13 Jun 12 16:52

personnes loggees sur Arthur : 7

peut aussi etre effectue par la commande Unix :

who | awk '{n++;print} END {print "personnes loggees sur Arthur :", n}'

2 : la fonction "system" La fonction "system (expression)" ex�ecute la commande
"expression" donn�ee sous forme de string :

Exemple :

plusieurs exemples d'ecritures sur le standard erreur.

BEGIN {

printf("\t Message : ")

getline message

print 1, message | "cat 1>&2"

system("echo '2 "message"' 1>&2")

print 3, message >"/dev/tty"

}

Message : And my blood is colder, closed the doors of my mind

1 And my blood is colder, closed the doors of my mind

2 And my blood is colder, closed the doors of my mind

3 And my blood is colder, closed the doors of my mind

Ce chapitre explique, �a l'aide d'exemples, tous les ingr�edients de programmation
en AWK.
La premi�ere section d�ecrit les patterns en d�etail. La seconde donne la description
des actions par les expressions, les a�ectations et les actions de tests et de boucles.
Les sections restantes couvrent la d�e�nition des fonctions, des sorties, des entr�ees et
comment des programmes awk peuvent int�eragir sur d'autres programmes.

Le �chier d'entr�ee choisi comme exemple dans cette partie sera un �chier texte

Fichier �elds :

Would you pay lifes pleasure

to see me

Does it heart for I want you to remain

I run your hair through, in another decade

Summerlands holds me in sumerian haze

Between the spaces, along the wall

appearing faces that disappear at dawn

Were getting closer, I can see the door

closer and closer

Kthulhu calls

forever remain, forever remain

2.1 Les patterns

Les patterns contrôlent l'ex�ecution des actions : quand un pattern est v�eri��e, il
matche la ligne concern�ee et l'action est ex�ecut�ee sur cette ligne. Cette partie d�ecrit
les six types de patterns et les conditions sous lesquelles elles sont v�eri��ees.

PATTERN EXEMPLE MATCHE

BEGIN BEGIN avant que toute entr�ee soit lue
END END apr�es que toute entr�ee soit lue

expression NF>5 matche les lignes o�u le nombre
de champs est sup�erieur �a 5

patterns sur chaines =closer= matche les lignes contenant la string
"closer"

compos�e de patterns NF>5 && =I= matche les lignes contenant plus de 5
champs et contenant la string "I"

intervalle NR==10, NR==20 matche les lignes de 10 �a 20

2.1.1 BEGIN et END

Les patterns BEGIN et END ne matchent aucune ligne en entr�ee. Les instructions
qui succ�edent le pattern BEGIN sont ex�ecut�ees avant toute lecture en entr�ee. De
fa�con similaire, les instructions qui succ�edent le pattern END sont ex�ecut�ees apr�es
toute lecture en entr�ee.
BEGIN et END ne se combinent avec aucun autre pattern.
Si plusieurs BEGIN apparaissent dans un programme, les actions associ�ees sont
ex�ecut�ees dans l'ordre auquel ils apparaissent dans le programme.

Exemple :

BEGIN {

print "on met en general les actions BEGIN en tete de fichier"

}

NF > 5

BEGIN {

printf("je suis toujours execute avant la premiere ligne d'entree\n")

}

on met en general les actions BEGIN en tete de fichier

je suis toujours execute avant la premiere ligne d'entree

Does it heart for I want you to remain

I run your hair through, in another decade

Summerlands holds me in sumerian haze

Between the spaces, along the wall

appearing faces that disappear at dawn

Were getting closer, I can see the door

2.1.2 Les expressions

Comme la plupart des langages de programmation, AWK est riche dans la descrip-
tion des op�erations arithm�etiques. Par contre, contrairement �a tous les langages,
AWK comprend des expressions sur les chaines de caract�eres.
Une string est une chaine de z�ero ou plus de z�ero caract�eres. Elles peuvent être
stock�ees dans des variables, ou peuvent apparaitre litt�eralement en tant que des
constantes telles que "Fredo". La chaine "" est appel�ee la chaine nulle. Le terme de
substring signi�e une s�equence de z�ero ou plus de z�ero caract�eres �a l'int�erieur d'une
string.
Toute expression peut être utilis�ee en tant qu'op�erande de tout op�erateur. Si une
expression a une valeur num�erique mais qu'un op�erateur requiert une string, alors la
valeur num�erique est automatiquement transform�ee en une string et r�eciproquement.
Les patterns les plus typiques sont les comparaisons entre les chaines ou nombres.
Ces op�erateurs sont list�es dans le tableau qui suit.
L'op�erateur � (tilde) sera expliqu�e dans la partie suivant.

OPERATEUR SIGNIFICATION

< plus petit que
<= plus petit ou �egal �a
== �egal �a
!= di��erent de
>= plus grand ou �egal �a
> plus grand que
� match�e par
!� non match�e par

Dans le cas d'une comparaison, si les 2 op�erandes sont num�eriques, une comparaison
num�erique est e�ectu�ee; sinon, tout op�erande num�erique est converti en une string
et alors les op�erandes sont compar�es en tant que string.
Les strings sont compar�ees caract�ere par caract�ere selon l'ordre d�e�ni par la ma-
chine, souvent selon les codes ASCII. Une chaine est dite inf�erieure �a une autre si
elle apparait avant l'autre selon cet ordre.

Exemple :

"Fumer" < "Stone"

"Fumee" < "Fumeur"

"Fumeur" < "fumeur"

Lorsqu'un pattern de comparaison est valid�ee sur la ligne courante, alors la ligne
courante est dite " match�ee ", c'est-�a-dire s�elect�ee.

2.1.3 Les patterns sp�eci�ques aux chaines

AWK permet de s�electer ou de matcher des chaines de caract�eres quelconques �a par-
tir d'expressions sp�eci�ques appel�ees EXPRESSIONS REGULIERES qui sont
tr�es souvent utilis�ees en Shell et sous �editeur.
Un pattern sur une chaine teste si la chaine contient e�ectivement la substring
match�ee par l'expression r�eguli�ere.
Le tableau ci-dessous donne les 3 grands types de patterns sp�eci�ques �a la recherche
de chaines de caract�eres :

TYPE SIGNIFICATION

=expression r�eguli�ere= Matche la ligne courante quand une substring
est match�ee par l'expression r�eguli�ere

expression�=expr.reg.= Matche la ligne courante si la chaine "expression"
contient une substring match�ee par l'expression r�eguli�ere

expression!�=expr.reg.= Matche la ligne courante si la chaine "expression"
ne contient pas une substring match�ee par l'expression r�eguli�ere

L'expression r�eguli�ere la plus simple est une suite de lettres ou de chi�res entre
slashs.

Exemple : =Fredo=
Ce pattern recherche sur chaque ligne les occurences de la substring "Fredo" et si
cette chaine est trouv�ee (même sous la forme FredoLeBon), alors la ligne est match�ee.

Remarque : = Fredo =
Ce pattern recherche la string "Fredo"; une ligne contenant FredoLeBon ne sera
donc pas match�ee.

Les deux autres types de patterns sp�eci�ques aux chaines utilisent un op�erateur
: l'op�erateur tilde : �

expression ~ /r/

expression !~ /r/

r represente une expression reguliere

L'op�erateur � signi�e "est match�e par" et l'op�erateur !� constitue son compl�ement.

Exemples :

$1 ~ /e/

# matche toutes les lignes dont le premier champ

# contient au moins un "e".

Does it heart for I want you to remain

Summerlands holds me in sumerian haze

Between the spaces, along the wall

appearing faces that disappear at dawn

Were getting closer, I can see the door

closer and closer

forever remain, forever remain

$1 !~ /e/

# matche toutes les lignes dont le premier champ

# ne contient pas de "e".

Would you pay lifes pleasure

to see me

I run your hair through, in another decade

Kthulhu calls

2.1.4 Les expressions r�eguli�eres

Une expression r�eguli�ere est une expression sp�eci�que contenant des symboles par-
ticuliers, qui permet la s�election de n'importe quel type de substring d'un texte par
exemple.
Elles sont toujours plac�ees entre slash : =r= comme nous l'avons d�ej�a vu dans la
partie pr�ec�edente. On les utilise aussi dans toutes les actions de conditions car nous
savons que les actions de conditions (tests et boucles) s'apparentent beaucoup aux
patterns.
Leur utilisation en tant que pattern se r�esume donc �a :

/r/

expression ~ /r/

expression !~ /r/

r represente une expression reguliere

Les caract�eres n,^,$,[ ],j,( ),�,+,? sont �a la base de toute construction d'expressions
r�eguli�eres; ils sont appel�es "m�etacaract�eres". Si on veut matcher l'un de ses symboles
dans une expression r�eguli�ere sous sa d�e�nition naturelle, on doit placer un antislash
devant (n$ matche le caract�ere $ par exemple).

1. n,^,$,. :
Le m�etacaract�ere n plac�e devant un autre m�etacaract�ere le transforme en un
caract�ere normal.
Le m�etacaract�ere n plac�e devant certains symboles sp�eciaux matche aussi une
s�equence de tabulations (voir le tableau ci-apr�es)

SEQUENCE SIGNIFICATION

nb backspace : supprime le dernier caract�ere d'une chaine
nf formfeed : nouvelle page
nn newline : nouvelle ligne
nr carriage return : retour �a la ligne
nt tabulation : cr�ee une tabulation de 10 espaces

nddd valeur octale ddd o�u ddd est trois nombres compris entre 0 et 7
nc tout caract�ere pris sous sa forme litt�erale

(except�e pour les caract�eres : n : nn et " : n")

Le m�etacaract�ere ^ matche le d�ebut d'une string.
Le m�etacaract�ere $ matche la �n d'une string.
Le m�etacaract�ere . matche un caract�ere quelconque et un seul.

Exemples :

^F matche un F au debut d'une chaine.

F$ matche un F en fin de chaine.

^F$ matche la chaine composee du simple caractere F.

^.$ matche toute chaine composee de un seul caractere.

... matche toute suite de trois caracteres.

\.$ matche le point en fin de chaine.

2. [ ] :
Une expression r�eguli�ere consistant en un groupe de caract�eres entour�e de
crochets est appel�ee classe de caract�eres; elle matche chacun des caract�eres �a
l'int�erieur des crochets.

Lorsqu'on veut matcher une s�erie de caract�eres qui se suivent d'apr�es l'ordre
d�e�nie par la machine, on peut utiliser le tiret.

La classe compl�ementaire d'une classe quelconque est d�enot�ee par le m�etacaract�ere
^ apr�es l'ouverture du crochet.

Exemples :

[AEO] matche chacun des caracteres A,E et O.

[^AEO] matche tous les caracteres sauf A,E et O.

^[^ABC] matche tout caractere en debut de chaine

sauf A,B et C.

^[^a-z-] matche toute chaine consistant en un

seul caractere sauf les minuscules et "-".

[a-zA-Z][0-9] matche toute chaine consistant en une lettre

minuscule ou majuscule suivie d'un nombre.

A l'int�erieur d'une classe de caract�ere, tout caract�ere admet sa signi�cation
normale except�e les caract�eres n , ^ plac�e en d�ebut et - plac�e entre 2 caract�eres.
Ainsi :

[.] matche le point ".".

^[^^] matche tout caractere en debut de chaine sauf l'accent ^.

3. ( ),j :
Il y a deux op�erateurs binaires sur les expressions r�eguli�eres : la concat�enation
et le ou.
Si r1 et r2 sont des expressions r�eguli�eres, alors :
r1jr2 matche toute string match�ee par r1 ou r2.
(r1)(r2) matche toute string de la forme xy o�u r1 matche x et r2 matche y.
Si r1 ou r2 ne contiennent pas d'op�erateur j, les parenth�eses peuvent être
omises.

Exemple :

/([a-zA-Z])(ee)/

# matche les substring contenant un caractere suivi

# de la chaine "ee".

/(forever|closer) (remain|and)/

# matche les lignes contenant l'un des deux

# mots presents dans la premiere parenthese

# suivi par l'un des deux mots presents dans

# l'autre parenthese.

to see me

Between the spaces, along the wall

Were getting closer, I can see the door

closer and closer

forever remain, forever remain



2.1. LES PATTERNS 43

4. +,�,? :
Ces op�erateurs unaires sont utilis�es pour sp�eci�er des r�ep�etitions dans des ex-
pressions r�eguli�eres.
(r)�matche toute chaine consistant en z�ero ou plus de z�ero substring cons�ecutives
match�ees par r.
(r)+matche toute chaine consistant en une ou plus de une substring cons�ecutives
match�ees par r.
(r)? matche la chaine nulle ou toute chaine match�ee par r.

Exemple :

F* : matche la chaine nulle ou F ou FF, etc...

FR*E : matche FE ou FRE ou FRRE, etc...

FR+E : matche FRE ou FRRE ou FRRRE, etc...

FRR*E : matche FRE ou FRRE ou FRRRE, etc...

FR?E : matche FE ou FRE.

[A-Z]+ : matche toute chaine de une ou plus de une lettre majuscule

(FR)+E : matche FRE, FRFRE, FRFRFRE, etc...

^[+-]?([0-9]+[.]?[0-9]*|[.][0-9]+)([eE][+-]?[0-9]+)?$

matche les nombres reels avec un signe optionnel

et un exposant optionnel.

Nous r�ecapitulons dans le tableau ci-dessous toutes les expressions r�eguli�eres que
l'on peut rencontrer et leur signi�cation.

EXPRESSION MATCHE

c le non m�etacaract�ere c
nc s�equence de tabulation ou le caract�ere litt�eral c
^ d�ebut de string
$ �n de string
. tout caract�ere

[c1c2:::] tout caract�ere c1, c2 ....
[^c1c2:::] tout caract�ere sauf c1, c2 ....
[c1-c2] tout caract�ere de c1 �a c2
[^c1-c2] tout caract�ere non de c1 �a c2
r1jr2 toute chaine match�ee par r1 ou r2

(r1)(r2) toute chaine xy o�u r1 matche x et o�u r2 matche y
les parenth�eses ne sont pas obligatoires si on n'utilise pas j

(r)� z�ero ou plus de z�ero chaines cons�ecutives match�ees par r
(r)+ une ou plus de une chaines cons�ecutives match�ees par r
(r)? la chaine nulle ou toute chaine match�ee par r
(r) toute chaine match�ee par r

2.1.5 Les compos�es de patterns

Un compos�e de patterns est un expression qui combine une s�erie de patterns par un
op�erateur logique : jj (OU), && (ET) et j (NON).
Un compos�e de pattern matche la ligne courante si l'expression �evalu�ee est vraie.

Exemple :

/(forever|closer)/ && /^[fc]/

closer and closer

forever remain, forever remain

2.1.6 Les intervalles

On d�e�nit un intervalle de patterns comme �etant deux patterns s�epar�es par une
virgule telle que :

pat1, pat2

Un intervalle de patterns matche chaque ligne entre une occurence de pat1 et la
prochaine occurence de pat2, incluse; pat2 peut matcher la même ligne que pat1
donnant ainsi en r�esultat de l'intervalle une seule ligne.

Exemples :

/see/, /I/

to see me

Does it heart for I want you to remain

Were getting closer, I can see the door

/I/, /see/

Does it heart for I want you to remain

I run your hair through, in another decade

Summerlands holds me in sumerian haze

Between the spaces, along the wall

appearing faces that disappear at dawn

Were getting closer, I can see the door

2.2 Les actions

Dans la structure d'un programme AWK :

PATTERN f ACTION g instruction
PATTERN f ACTION g instruction

...............................

les patterns d�eterminent quand une action doit s'ex�ecuter.
Parfois, une action peut être tr�es simple : print ... ou elle peut être une a�ec-

tation; souvent, elle regroupe une s�erie de lignes contenant des fonctions, accolades
et autres �el�ements du vocabulaire non terminal qui d�e�nissent toute la richesse du
langage AWK.

Le tableau ci-dessous donne tous les �el�ements du vocabulaire de haut niveau
intervenant dans une action :

ACTIONS

expressions avec des constantes, variables, a�ectations, appel de fonctions, etc..
print(format,liste d'expression)
printf(format,liste d'expression)
if (expression) instructions
if (expression) instructions else instructions

while (expression) instruction
for (expression;expression;expression) instructions
for (variable in tableau) instruction
do instructions while (expression)
break
continue
next
exit
exit expression
f expression g

2.2.1 Les Expressions

Les expressions les plus simples sont des blocs primaires constitu�es de constantes,
de variables, de tableaux, d'appels �a des fonctions et de variables pr�e-d�e�nies telles
que les s�eparateurs.

Les op�erateurs sur les expressions combinent les expressions entre elles et ils
entrent dans cinq grandes cat�egories : arithm�etique, comparaison, logique, condi-
tionelle et a�ectation.

En�n, les fonctions pr�ed�e�nies et les fonctions que l'on peut construire utilisent
les expressions et repr�esentent aussi une grande classe d'expressions.

� Les constantes, variables, variables pr�e-d�e�nies
Il existe deux types de constantes, les chaines et les nombres.
Une constante chaine est cr�e�ee en entourant la chaine de quotes : " telle que
"pleasure ", "holds", "me ", "for", "", "?".
Une constante num�erique peut être un entier (2000) ou un r�eel sous toutes les formes
suivantes :
2000.01
0.07E-1 ou 0.07e-1
1e6, 1.00E6, 10e5, 0.1e7 et 1000000 sont identiques
Tous les nombres ont la pr�ecision maximale d�ependant de la machine.

� Les variables
Il y a deux types de variables sous AWK :

1. les variables d�e�nies par l'utilisateur : ce sont toutes les s�equences de lettres,
chi�res et soulign�es qui ne commencent pas par un chi�re et qui ne font pas
partie du vocabulaire non terminal du langage AWK.
Puisque le type d'une variable n'est pas d�eclar�e, AWK d�etermine son type
de part le contexte du programme. Ainsi, AWK convertit la valeur d'une
chaine en un nombre ou en une chaine suivant la fa�con dont on veut utiliser
la variable.

2. les variables pr�e-d�e�nies : il existe sous AWK des variables globales qui sont
�x�ees au d�epart et que l'on ne peut utiliser al�eatoirement. Par contre, la valeur
de ces variables peut être modi��ee par l'utilisateur tout en restant coh�erent
avec sa d�e�nition.

Le tableau ci-dessous liste la forme litt�erale, la d�e�nition et la valeur par d�efaut
de chacune de ces variables.

VARIABLE DEFINITION DEFAUT

ARGC nombre d'arguments de la ligne de commande -
ARGV tableau d'arguments de la ligne de commande -
FILENAME nom du �chier d'entr�ee courant -
FNR nombre d'enregistrement du �chier courant -
FS controle le s�eparateur de champ en entr�ee " "
NF nombre de champ de l'enregistrement courant -
NR nombre d'enregistrements lus -
OFMT format de sortie pour les nombres "%.6g"
OFS s�eparateur de sortie " "
ORS s�eparateur des enregistrements "nn"
RLENGTH longueur d'une chaine match�ee par la fonction "match" -
RS controle le s�eparateur d'enregistrement en entr�ee "nn"
RSTART d�ebut de la chaine match�ee par la fonction "match" -
SUBSEP s�eparateur d'indice "n034"

Exemple :

BEGIN {

OFS=" "

print "nombre d'arguments :", ARGC "\n"

for (c in ARGV) print c,ARGV[c]

printf("\n")

for (i=0;i<ARGC;i++) printf(" %s", ARGV[i])

printf("\n\n")

printf("NOM du fichier1 : %s\n", FILENAME)

}

{print "FNR=" FNR,"NR=" NR,$0}

END {

printf("NOM du fichier2 : %s\n", FILENAME)

}

nombre d'arguments : 3

2 fields

0 awk

1 fields

awk fields fields

NOM du fichier1 : fields

FNR=1 NR=1 Would you pay lifes pleasure

FNR=2 NR=2 to see me

FNR=3 NR=3 Does it heart for I want you to remain

FNR=4 NR=4 I run your hair through, in another decade

FNR=5 NR=5 Summerlands holds me in sumerian haze

FNR=6 NR=6 Between the spaces, along the wall

FNR=7 NR=7 appearing faces that disappear at dawn

FNR=8 NR=8 Were getting closer, I can see the door

FNR=9 NR=9 closer and closer

FNR=10 NR=10 Kthulhu calls

FNR=11 NR=11 forever remain, forever remain

FNR=1 NR=12 Would you pay lifes pleasure

FNR=2 NR=13 to see me

FNR=3 NR=14 Does it heart for I want you to remain

FNR=4 NR=15 I run your hair through, in another decade

FNR=5 NR=16 Summerlands holds me in sumerian haze

FNR=6 NR=17 Between the spaces, along the wall

FNR=7 NR=18 appearing faces that disappear at dawn

FNR=8 NR=19 Were getting closer, I can see the door

FNR=9 NR=20 closer and closer

FNR=10 NR=21 Kthulhu calls

FNR=11 NR=22 forever remain, forever remain

NOM du fichier2 : fields

On peut additionner des arguments �a ARGV ou modi�er son contenu; ARGC
peut aussi être modi��e. A chaque �n de �chier d'entr�ee, awk traite l'�el�ement
suivant non nul de ARGV (jusqu'�a la valeur de ARGC-1) comme le nom du
prochain �chier d'entr�ee. Ainsi, mettre un �el�ement de ARGV �a z�ero signi�e
qu'il ne sera pas trait�e comme un �chier d'entr�ee. Le nom "-" peut être utilis�e
pour l'entr�ee standard. nous voyons un exemple dans la partie Int�eraction
avec les autres programmes.

� les op�erateurs arithm�etiques, logiques et d'a�ectation
AWK permet la combinaison des expressions par les op�erateurs.
Nous listons les di��erents op�erateurs dans le tableau ci-dessous.

OPERATION OPERATEURS EXEMPLE SIGNIFICATION

a�ectation = += -= �= x� = 2 x = x � 2
/= %= ^=

conditionnel ? : x?y:z si x est vrai alors y sinon z
logique OU jj x jj y 1 si x ou y sont vrais

0, sinon
logique ET && x && y 1 si x et y sont vrais

0, sinon
�el�ement de tableau in i in tab 1 si tab[i] existe, 0 sinon
matching ~ !~ $1 ~ /x/ 1 si le premier champ

contient un x, 0 sinon
relationel < <= == x == y 1 si x est �egal �a y

!= >= > 0,sinon
concat�enation "Bad" " Deal" "Bad Deal"; il 'y a pas

d'op�erateur explicite
de concat�enation

somme, soustraction + - x + y somme de x et de y
mult.,div.,modulo � / % x % y reste de la division de x par y
plus et moins unaire + - -x oppos�e de x
logique NON j j$1 1 si $1 n'est pas z�ero ou ""

0, sinon
puissance ^ x^y xŷ
incr�ement ++ { ++x, x++ additionne 1 �a x
d�ecr�ement
champ $ $i+1 valeur du i�eme champ

additionn�e de 1
groupement ( ) ($i)++ additionne 1 �a la valeur

du i�eme champ

Op�erateurs arithm�etiques : Ce sont +,-,�,/,% et ^ qui s'e�ectuent uniquement
sur des nombres. Toutes les op�erations arithm�etiques sont faites en pr�ecision maxi-
male. Si une variable sous forme de chaine constitue l'un des membres de l'op�erateur,
elle est convertie en sa valeur qui doit être un nombre.

Op�erateurs logiques : Les op�erateurs && (ET), jj (OU) et j (NON) sont uitlis�es
pour cr�eer des expressions logiques en combinant d'autres expressions. Une expres-
sion logique a une valeur de 1 si elle est vraie ou de 0 dans le cas contraire. AWK
�evalue les op�erations de gauche �a droite et ne fait pas de travaux superus.
Exemple : pour expr1 && expr2, expr2 n'est pas �evalu�e si expr1 est faux.

Expressions de condition : Une expression de condition a la forme :
expr1 ? expr2 : expr3

expr1 est �evalu�ee; si elle est vraie, c.a.d. non nulle ou non �egale �a "", la valeur de
l'expression est �egale �a la valeur de expr2, sinon la valeur de l'expression est �egale �a
la valeur de l'expression expr3.

Op�erateurs d'a�ectation : Il y a 7 types d'a�ectations dont la plus simple est de
la forme variable = expression.
Les 6 autres op�erateurs d'a�ectation sont +=,-=,�=,/=,%= et ^= qui repr�esentent
une simpli�cation d'utilisation du premier type d'op�erateur d'a�ectation pour cer-
tains cas particuliers.
Exemple : age=age-10 peut s'�ecrire plus simplement age-=10 ce qui fait rajeunir.

Op�erateurs d'incr�ementation et de d�ecr�ementation : l'expression n=n+1 est plus
commod�ement repr�esent�ee par n++ ou ++n. La forme pr�e�x�ee ++n incr�emente
n avant de d�elivrer sa valeur alors que la forme post�x�ee n++ incr�emente n apr�es
avoir d�elivr�e sa valeur.
Exemple:

BEGIN {

printf("\t\tParodie de la vie \n")

n=0

i=n++; printf("\t j'ai %d ans et je viens de naitre\n",n*10^i)

++n

j=++i; printf("\t j'ai %d ans et je suis un jeune initie\n",n*10^j)

printf("\t j'ai %d ans et je suis parvenu a la sagesse\n",(j+n)*10^n)

}

Parodie de la vie

j'ai 1 ans et je viens de naitre

j'ai 20 ans et je suis un jeune initie

j'ai 300 ans et je suis parvenu a la sagesse



52 CHAPTER 2. LE LANGAGE AWK EN D�ETAIL

Op�erateurs sur les chaines : il n'y a qu'un type d'op�erateur sur les chaines :
la concat�enation. Cet op�erateur n'est pas explicite; les expressions sont cr�e�ees en
�ecrivant des constantes, variables, champs, �el�ements de tableaux, r�esultats de fonc-
tion ou autres expressions les unes �a la suite des autres

{ print NR ":" $0 }

1:Would you pay lifes pleasure

2:to see me

3:Does it heart for I want you to remain

4:I run your hair through, in another decade

5:Summerlands holds me in sumerian haze

6:Between the spaces, along the wall

7:appearing faces that disappear at dawn

8:Were getting closer, I can see the door

9:closer and closer

10:Kthulhu calls

11:forever remain, forever remain

Chaines en tant qu'expressions r�eguli�eres : Dans beaucoup de nos exemples pr�ec�edents,
le membre de droite de l'op�erateur ~ ou !~ �etait une expression r�eguli�ere entour�ee
de slashs. En fait, toute expression peut être utilis�ee dans le membre de droite de
ces deux op�erateurs. AWK �evalue l'expression, convertit la chaine si n�ecessaire et
interpr�ete la chaine comme une expression r�eguli�ere.
Exemple : imprime les lignes o�u apparaissent les mots see et closer.

BEGIN { a_trouver = "(see|closer)" }

$0 ~ a_trouver

to see me

Were getting closer, I can see the door

closer and closer

Une expression r�eguli�ere peut être pass�ee dans une expression de variable et ainsi
peut aussi être concat�en�ee.
Exemple : reconnaissance d'un nombre r�eel

BEGIN {

signe="[+-]?"

decimal="[0-9]+[.]?[0-9]*"

fraction="[.][0-9]+"

exposant="([eE]" signe "[0-9]+)?"

nombre="^" signe "(" decimal "|" fraction ")" exposant "$"

}

$0 ~ nombre

On peut donc construire une expression r�eguli�ere entre quotes comme on con-
struit une expression r�eguli�ere entre slashs pour les patterns; il y a une seule ex-
ception pour les m�etacaract�eres; lorsque l'on veut matcher le m�etacaract�ere sous sa
forme normale, on doit ajouter un extra slash dans la forme quot�ee :

$0 ~ /(\+|-)[0-9]+/

$0 ~ "(\\+|-)[0-9]+"

La suite donne quelques exemples d'utilisation d'expressions r�eguli�eres. Pour
cela, nous modi�ons l�eg�erement le �chier "�elds" en ins�erant en d�ebut et en �n de
ligne une s�erie quelconque de blancs. Le �chier �elds a alors cette allure

Would you pay lifes pleasure

to see me

Does it heart for I want you to remain

I run your hair through, in another decade

Summerlands holds me in sumerian haze

Between the spaces, along the wall

appearing faces that disappear at dawn

Were getting closer, I can see the door

closer and closer

Kthulhu calls

forever remain, forever remain

# PATTERNS qui teste si une chaine contient

# une substring en utilisant les expressions regulieres.

BEGIN {print " Les vers qui contiennent le mot 'see' et qui ne"

print " contiennent pas le mot 'I'"}

$0 ~ /see/ && $0 !~ /I/ {print NR ," " ,$0}

Les vers qui contiennent le mot "see" et qui ne

contiennent pas le mot "I"

2 to see me

# Exemples en utilisant des metacaracteres

#On affiche les vers commencant par une majuscule

BEGIN {

print " Les vers commencant par une majuscule :"

}

$0~/^( )*[A-Z]/ {print NR, " ",$0} # prise en compte des blancs

# en debut de ligne

Les vers commencant par une majuscule :

1 Would you pay lifes pleasure

3 Does it heart for I want you to remain

4 I run your hair through, in another decade

5 Summerlands holds me in sumerian haze

6 Between the spaces, along the wall

8 Were getting closer, I can see the door

10 Kthulhu calls

#On affiche les vers se terminant par "e"

BEGIN {

print " Les vers se terminant par un \"e\" :"

}

$0~/e( )*$/ {print NR," ",$0} # prise en compte des blancs

# en fin de ligne

Les vers se terminant par un "e" :

1 Would you pay lifes pleasure

2 to see me

4 I run your hair through, in another decade

5 Summerlands holds me in sumerian haze

#On affiche les champs contenant la chaine "ces"

BEGIN {

print " Les champs qui contiennent la chaine \"ces\" :"

}

$0~/ces/ {

for (i=1;i<=NF;i++)

{

if ($i~/ces/) print $i "\n"

}

}

Les champs qui contiennent la chaine "ces" :

spaces,

faces

#On affiche les lignes ne commencant pas par

#les lettres B,A,D,b,a,d :

BEGIN {

print "On affiche les lignes ne commencant pas"

print "par les lettres B,A,D,b,a,d :"

}

$0 !~ "^( )*[BAD]|^( )*[bad]" {print}

On affiche les lignes ne commencant pas

par les lettres B,A,D,b,a,d :

Would you pay lifes pleasure

to see me

I run your hair through, in another decade

Summerlands holds me in sumerian haze

Were getting closer, I can see the door

closer and closer

Kthulhu calls

forever remain, forever remain

� Les Fonctions pr�e-d�e�nies en AWK
AWK a des fonctions d�e�nies dans son propre langage qui sont les fonctions arithm�etiques
et les fonctions sur les chaines.

les fonctions arithm�etiques : ces fonctions peuvent être utilis�ees en tant qu'expression
primaire dans toutes les expressions d�e�nies dans cette partie. Elles sont list�ees dans
le tableau ci-dessous :

FONCTION VALEUR RETOURNEE
cos(x) cosinus de x, avec x en radians
sin(x) sinus de x, avec x en radians
atan2(y,x) arctangente de y/x retourn�ee dans l'intervalle -� et �
sqrt(x) racine carr�ee de x
log(x) logarithme n�ep�erien de x
exp(x) exponentielle de x
int(x) partie enti�ere de x
rand() nombre au hasard r, 0 � r < 1
srand(x) x d�e�nit le point de d�epart dans la g�en�eration au hasard rand()

srand() fait d�ebuter le g�en�erateur selon l'heure

atan2(0,-1) donne �.
exp(1) donne e.
le logarithme d�ecimal se d�e�nit par log(x)= log(10).
La fonction rand() retourne un nombre de pr�ecision maximale selon un pseudo-
hasard. En e�et, la fonction srand(x) d�e�nit �a partir de x le d�ebut de la g�en�eration
au hasard. Si on appelle srand() sans param�etre, alors la g�en�eration commencera �a
partir de l'heure. Si on n'appelle pas srand, rand commencera toujours �a la même
valeur, chaque fois que le programme est appel�e.
Exemple : faites tourner deux fois les deux programmes suivants :

BEGIN {srand(43) BEGIN {srand()

x=rand();print int(100*x) x=rand();print int(100*x)

x=rand();print int(100*x) x=rand();print int(100*x)

x=rand();print int(100*x) x=rand();print int(100*x)

x=rand();print int(100*x) x=rand();print int(100*x)

} }

9 56

4 34

46 19

63 3

9 58

4 38

46 20

63 75

les fonctions AWK sur les chaines : AWK permet l'utilisation des fonctions pr�ed�e�nies
qui op�erent sur les chaines. Ces fonctions sont list�ees dans le tableau ci-dessous.

FONCTION ROLE

length(s) retourne le nombre de caract�ere de s
index(s,t) retourne la premi�ere position de la chaine t dans s

ou 0 si t n'est pas pr�esent
match(s,r) teste si s contient une substring match�ee par r

retourne l'index ou 0
valeurs dans RSTART et RLENGTH

sprintf(fmt,list-expr) retourne list-expr formatt�ee selon le format fmt
substr(s,p) retourne le su�xe de s commen�cant �a la position p
substr(s,p,n) retourne la substring de s de longueur n

et commen�cant �a la position p
split(s,a) �eclate s dans un tableau a selon le s�eparateur FS

retourne le nombre d'�el�ements du tableau a
split(s,a,fs) �eclate s dans un tableau a suivant le s�eparateur fs

retourne le nombre d'�el�ements du tableau a
sub(r,s) substitue par s la plus �a gauche et la plus longue

substring dans $0 match�ee par r
retourne le nombre de substitutions faites

sub(r,s,t) substitue par s la plus �a gauche et la plus longue
substring dans t match�ee par r
retourne le nombre de substitutions faites

gsub(r,s) substitue par s toutes les chaines
match�ees par r dans $0
retourne le nombre de substitutions faites

gsub(r,s,t) substitue par s toutes les chaines
match�ees par r dans t
retourne le nombre de substitutions faites

r repr�esente une expression r�eguli�ere;
s et t sont des strings; n et p sont des entiers.

- La fonction length(s) retourne la longueur de la chaine s en tenant compte des
blancs.
Exemple :

BEGIN {chaine="Tous ces vieux en longs habits de Cours !?"

print length(chaine)}

44

- La fonction index(s,t) retourne la position la plus �a gauche de la chaine t dans
la chaine s ou 0 si t n'apparait pas. Le premier caract�ere d'une chaine a la position
1 et les blancs sont pris en compte.
Exemple :

BEGIN {x=index("Les nana ben's", "na");print x}

5

- La fonction match(s,r) cherche la substring la plus longue et la plus �a gauche
qui est match�ee par l'expression r�eguli�ere r dans la chaine s. Elle retourne l'index
o�u la substring commence ou 0 si r ne matche rien. Elle impl�emente aussi les deux
variables RSTART �a la valeur de l'index et RLENGTH �a la longueur de la substring
match�ee.
Exemple :

{if (match($0,/,/)>0)

{

printf("la virgule apparait a la ligne %d, a la position %d\n",NR,RSTART)

}

}

la virgule apparait a la ligne 4, a la position 24

la virgule apparait a la ligne 6, a la position 19

la virgule apparait a la ligne 8, a la position 20

la virgule apparait a la ligne 11, a la position 15

- La fonction sprintf(fmt,expr1; expr2; :::; exprn) retourne sans l'imprimer une
chaine contenant expr1; expr2; :::; exprn formatt�ee selon le format d�e�ni par l'expression
fmt et en se r�ef�erant aux sp�eci�cations de la fonction "printf".
Exemple :

{x=sprintf("%11s %7s", $1, $2);print x}

Would you

to see

Does it

I run

Summerlands holds

Between the

appearing faces

Were getting

closer and

Kthulhu calls

forever remain,

- La fonction substr(s,p) retourne le su�xe de s qui commence �a la position p.
Si substr(s,p,n) est utilis�e, alors seulement les n premiers caract�eres sont retourn�es.
Si le su�xe est de longueur inf�erieure �a n, alors tout le su�xe est retourn�e.
Exemple : supprime le "s" de "lifes" de la ligne 1 et concat�ene le r�esultat avec
d'autres chaines.

NR==1 {print substr($4,1,4), "is", $5}

life is pleasure

- La fonction split(s,a,fs) �eclate la chaine s dans le tableau a selon le s�eparateur
fs et retourne le nombre d'�el�ements ins�er�es dans le tableau a. Cette fonction sera
mieux d�ecrite dans la partie r�eserv�ee aux tableaux.

- La fonction sub(r,s,t) cherche d'abord la substring la plus �a gauche et la plus
longue match�ee par r dans la chaine cible t; si une substring est trouv�ee, elle est
alors substitu�ee par la string s. Elle retourne le nombre de substitutions e�ectu�ees.
La fonction sub(r,s) est synonyme de sub(r,s,$0).
La fonction gsub(r,s,t) est similaire sauf qu'elle e�ectue toutes les substitutions sur
la chaine t �a chaque fois qu'elle trouve une substring match�ee par r.
Dans la string qui substitue, tout caract�ere apparait sous sa forme litt�erale.
Exemple :

comparez les deux fonctions sub et gsub :

NR<5 {sub(/.ou/,"OU");print}

OUld you pay lifes pleasure

to see me

Does it heart for I want OU to remain

I run OUr hair through, in another decade

NR<5 {gsub(/.ou/,"OU");print}

OUld OU pay lifes pleasure

to see me

Does it heart for I want OU to remain

I run OUr hair thOUgh, in another decade

Le programme qui suit recupere les premiers champs des lignes tout

en supprimant ceux qui apparaissent deux fois : on concatene

tous les premiers champs pour creer une expression reguliere

a chaque fois qu'un nouveau champ reapparait.

Pour le test, on rajoutera la ligne suivante au fichier "fields" :

to escape your body from this

Funeral end

NR==1 {champ="(" $1 ")"}

NR!=1 && $1!~champ {gsub(/\)/,"",champ);champ=champ "|" $1 ")"}

END {

print champ

}

(Would|to|Does|Funeral|I|Summerlands|Between|appearing|

Were|closer|Kthulhu|forever)

� Les actions de tests et les boucles
La syntaxe des tests et des boucles est la même que celle du langage C.
Nous regroupons tous les types d'actions de tests et les boucles dans la liste ci-
dessous :

f instruction g
groupement d'instructions

if (expression) instruction
si expression est vraie, ex�ecution de instruction

if (expression) instruction1 else instruction2

si expression est vraie, ex�ecution de instruction1

sinon ex�ecution de instruction2

while (expression) instruction
tant que expression est vraie, ex�ecution de instruction

do instruction while (expression)
ex�ecute une fois instruction; apr�es, idem que while

for (expression1; expression2; expression3) instruction
�equivalent �a :
expression1; while (expression2) finstruction; expression3g

for (variable in tableau) instruction
ex�ecute instruction avec la valeur de variable

�egale �a chaque indice pris dans un ordre quelconque de tableau

break
sort imm�ediatement d'une boucle while, for ou do

continue
force le commencement de la prochaine it�eration
dans une boucle while, for ou do

next
force le commencement de la prochaine it�eration
dans le programme principal

exit
exit expression

force le programme �a ex�ecuter les actions END s'il en existe
sinon sort du programme
retourne expression

Il y a donc deux instructions qui modi�ent le cycle normal des boucles : break
et continue.
- l'instruction "break" provoque une sortie forc�ee d'une boucle while, for ou do.
- l'instruction "continue " provoque le commencement de l'it�eration suivante : le
programme ex�ecute alors l'expression de test du while ou du do ou expression3 du
for.

Les instructions next et exit controlent le d�eroulement d'un programme AWK
- l'instruction "next" provoque le passage �a la prochaine ligne du programme et
l'ex�ecution du programme reprend �a la prochaine instruction patternfactiong.
- dans les actions END, l'instruction "exit" provoque la �n du programme. Dans
toute autre action, cela cause la �n de toute lecture d'entr�ee et les actions END sont
alors ex�ecut�ees.

� Les Tableaux
AWK permet la cr�eation de tableaux qui n'ont pas besoin d'être d�eclar�es et dont le
nombre d'�el�ements n'a pas besoin d'être sp�eci��e.
On peut ainsi cr�eer un tableau comme sous n'importe quel type de langage : tableau
de chaines ou de nombres indic�e par des nombres (Ex : x[NR] = $0).
En fait, les indices d'un tableau sont d�e�nis comme des chaines ce qui constitue
une grande di��erence vis �a vis des langages usuels. On parle alors de TABLEAUX
ASSOCIATIFS. Ainsi, puisque la valeur des chaines 1 et "1" est la même, il revient
au même d'�ecrire tab[1] et tab["1"].On remarque par contre que tab[01] est di��erent
de tab[1] et tab[disappear] est di��erent de tab["disappear"]; dans le premier cas,
l'indice sera la valeur de la variable disappear et comme c'est une constante, les
�el�ements associ�es �a tab[disappear] seront accumul�es dans l'�etat disparu tab[""].

Exemple : nombre d'occurence des mots commen�cant par une voyelle minuscule.
tri sur le r�esultat.

{for (i=1;i<=NF;i++)

{

gsub(/,/,"",$i) # suppression de la virgule

++tab[$i]

}

}

END {printf("nombre d'occurence des mots\n")

printf("commencant par une voyelle minuscule\n")

for (mot in tab )

{

if (mot~/^[aeiouy]/)

{ printf("%s:%d\n",mot,tab[mot])|"sort" }

}

close("sort")

}

nombre d'occurence des mots

commencant par une voyelle minuscule

along:1

and:1

another:1

appearing:1

at:1

in:2

it:1

you:2

your:1

Le programme pr�ec�edent utilise une forme de l'instruction "for" qui boucle sur
tous les indices du tableau :

for (variable ) in tab

instructions

La boucle ex�ecute les instructions avec la variable prenant comme valeur chaque
indice du tableau. L'ordre dans lequel les variables prennent les valeurs des indices
est al�eatoire.
Les r�esultats sont impr�evisibles si de nouveaux �el�ements sont ajout�es dans le tableau
par les instructions.
On peut d�eterminer si un indice apparait dans un tableau ou non par l'expression

indice in tab

Cette expression prend la valeur 1 si tab[indice ] existe d�ej�a, et 0 sinon.

Exemple : cherche si "Death" est dans le tableau tab cr�e�e pr�ec�edemment.

{for (i=1;i<=NF;i++)

{

gsub(/,/,"",$i)

++tab[$i]

}

}

END {

if ("Death" in tab)

{

printf("\tO flammes du sommeil sur un visage d'ange\n")

printf("\tEt sur toutes les nuits et sur tous les visages\n")

}

else

{

printf("\tSilence. Le silence eclatant de ses reves\n")

printf("\tQui enivrent d'ouragans une ame delivree\n")

}

}

Silence. Le silence eclatant de ses reves

Qui enivrent d'ouragans une ame delivree

L'instruction delete d�etruit un tableau avec la syntaxe :
delete tab [indice]

Exemple :

for (i in zombi) {delete zombi[i]} # le tableau zombi est mort

La fonction split (str,tab,fs) �eclate la valeur de la chaine str en champs selon le
s�eparateur fs et place chacun des champs dans le tableau tab. Le nombre de champs
produits ou le nombre d'�el�ements de tab est retourn�e par la fonction.

Exemple : �eclatement de la chaine "cheveux en petard" selon le s�eparateur e.

BEGIN {chaine="cheveux en petard"

max=split(chaine, tab, "e")

for (i=1;i<=max;i++) print tab[i]

}

ch

v

ux

n p

tard

On peut aussi cr�eer des tableaux multidimensionels en AWK qui sont en fait des
simulations utilisant des tableaux �a une dimension. Bien que l'on puisse utiliser des
indices tels que i,j ou s,p,q,r, awk concat�ene les indices (en glissant un s�eparateur
entre eux) pour former un unique indice qui n'a plus rien �a voir avec l'�ecriture de
l'utilisateur.
L'exemple :

for (i=1;i<=10;i++)

for (j=1;j<=10;j++)

tab[i,j]=0

cr�ee un tableau de 100 �el�ements dont les indices ont la forme :
1,1

1,2 ......

En fait, les indices sont stock�es comme des chaines qui ont la forme :
1 SUBSEP 1

1 SUBSEP 2 ......

La valeur de la variable pr�e-construite SUBSEP est par d�efaut "\034"

Pour tester si une entit�e appartient �a un tableau multidimensionel, on peut �ecrire :
if ((i,j) in tab) ...

Pour r�ecup�erer tous les �el�ements du tableau, il su�t de faire :
for (k in tab) ...

Exemple :

BEGIN {

for (i=1;i<=2;i++)

for (j=1;j<=2;j++)

tab[i,j]=i+j

for (k in tab)

{

split(k,x,SUBSEP)

print "tab[" k "]=",tab[k] " tab[" x[1] "," x[2] "]=",tab[k]

}

}

tab[11]= 2 tab[1,1]= 2

tab[12]= 3 tab[1,2]= 3

tab[21]= 3 tab[2,1]= 3

tab[22]= 4 tab[2,2]= 4

2.2.2 Cr�eer une Fonction

En plus des fonctions pr�e-d�e�nies, AWK autorise l'utilisateur �a cr�eer ses propres
fonctions. Une fonction se d�e�nit de la fa�con suivante :

function nom(liste de param�etres)
f
instructions

g

La forme g�en�erale d'un programme AWK devient maintenant une s�equence de
pattern-action suivie des d�e�nitions des fonctions cr�e�es par l'utilisateur.
La liste des param�etres est une s�equence de variables s�epar�ees par des virgules; �a
l'int�erieur du corps de la fonction, ces variables r�ef�erent aux arguments pass�es en
param�etre lors de l'appel de la fonction.
Le corps de la fonction peut contenir l'instruction return qui retourne la valeur
d'une expression qui peut servir dans le niveau sup�erieur appelant. L'expression est
optionnelle.

return expression

A l'int�erieur d'une fonction, les param�etres sont des variables locales; elles ne
durent que le temps d'ex�ecution de la fonction et sont ensuite d�epil�ees. Mais
toutes les autres variables sont globales; si une variable n'est pas dans la liste des
param�etres, elle peut r�ef�erencer �a une variable existant dans le programme appelant
et ainsi on peut en �ecraser son contenu.

Exemple : �evaluation �a l'aide du programme pr�ec�edent des mots qui apparaissent
le plus de fois dans le �chier "�elds".

{for (i=1;i<=NF;i++)

{

gsub(/,/,"",$i)

++tab[$i]

}

}

END {

for (c in tab)

{maxi=max(tab[c],maxi)}

printf(" Mots les plus frequents dans ce dedale obscur\n\t\t\"")

for (c in tab)

{if (tab[c]==maxi) {printf("%s ", c)}}

printf("\b\"\n\t C'est la parole des Fields\n")

}

function max(a, b)

{

return a > b ? a : b

}

Mots les plus frequents dans ce dedale obscur

"the I closer remain"

C'est la parole des Fields

2.2.3 Les sorties

Les fonctions print et printf g�en�erent les sorties sous AWK. "print" est utilis�ee
comme nous l'avons d�ej�a vu pour des �ecritures simples alors que "printf" demande
un format d'�ecriture que nous d�etaillons dans cette partie. Les sorties peuvent aussi
bien être redirig�ees vers un �chier, un pipe que vers le standard output. Nous listons
dans le tableau qui suit toutes les fa�cons possibles de g�en�erer des sorties.

print
�ecrit $0 sur le standard output.

print expression, expression ...

�ecrit les expressions s�epar�ees par OFS, termin�ees par ORS.
print expression, expression ... > �chier

�ecrit dans le �chier "�chier" en l'ouvrant s'il n'existe pas
ou en l'�ecrasant s'il existe d�ej�a.

print expression, expression ... >> �chier

�ecrit �a la suite du �chier "�chier".
print expression, expression ... j commande

�ecrit vers l'entr�ee de commande.

printf(format,expression, expression ...)
printf(format,expression, expression ...) > �chier

printf(format,expression, expression ...) >> �chier

printf(format,expression, expression ...) j commande

idem que pour "print" en sp�eci�ant un format �a chaque fois.

close(�chier), close(commande)
clôture la connection entre "print" et �chier ou commande

fermeture des pipes.

system(commande)
ex�ecute commande; la valeur retourn�ee est celle de commande.

� La fonction "print" a deux formes :

print expr1, expr2, ... , exprn

print(expr1, expr2, ... , exprn)

qui ont le même rôle : �ecriture de chaque expression s�epar�ee par le s�eparateur de
champ de sortie (OFS : output �eld separator) et suivie du s�eparateur d'enregistrement
de sortie (ORS : output record separator).
L'instruction print est �equivalente �a print $0 : elle �ecrit l'enregistrement $0 en
sortie.
L'instruction print "" �ecrit une ligne blanche en sortie.

� Les s�eparateurs de sortie :
Il existe deux s�eparateurs de sortie : OFS qui d�e�nit le s�eparateur entre chaque
champ et ORS qui d�e�nit le s�eparateur apr�es un enregistrement (un enregistrement
�etant une ligne de champs).
Exemple :

BEGIN { FS="faces"

OFS=", "

ORS="\nThat's the last exit for the lost ..........\n" }

$0 ~ FS {

max=split($1,T21," ")

gsub(/ing/,"ed",T21[1])

gsub("^a","A",T21[1])

print "The seven keys","\nIn front of my " FS ,"\n" T21[1] "."

}

The seven keys,

In front of my faces,

Appeared.

That's the last exit for the lost ..........

� La fonction "printf" est similaire �a celle du langage C. De même que "print", il y
a deux formes d'�ecriture de son expression :

printf format, expr1, expr2, ... , exprn

printf(format, expr1, expr2, ... , exprn)

Le format est une expression qui contient �a la fois du texte normal et des sp�eci�cations
sur le format de sortie des expressions pass�ees en arguments �a la fonction "printf".

Chaque sp�eci�cation commence par un %, se termine par un caract�ere qui
d�etermine la conversion requise et peut inclure trois autres modi�cateurs :

- justifie a gauche l'expression.

largeur determine l'emplacement en nombre de cases

ou sera placee la chaine ou le nombre.

.nombre taille maximale de la chaine ou nombre

de chiffres apres la virgule.

Nous donnons dans les tableaux qui suivent la liste des caract�eres qui interviennent
dans l'�ecriture des formats et des exemples de leur utilisation.

CARACTERE FORMAT D'ECRITURE

c caract�ere ASCII
d partie enti�ere d'un nombre
e [-]d.ddddddE[+-]dd
f [-]ddd.dddddd
g e ou f selon ce qui est le plus court;

supprime les z�eros en trop
o nombre octal non sign�e
s chaine
x nombre hexad�ecimal non sign�e
% �ecrit un %, aucun argument n'�etant utilis�e.

FORMAT : fmt EXPRESSION : $1 printf(fmt, $1)

%c 97 a

%d 97.5 97

%5d 97.5 97

%e 97.5 9.750000e+01
%f 97.5 97.500000

%7.2f 97.5 97.50

%g 97.5 97.5

%.6g 97.5 97.5

%o 97 141

%06o 97 000141

%x 97 61

|%s| Death |Death|

|%10s| Death | Death|

|%-10s| Death |Death |

|%.3s| DEATH |DEA|

|%10.3s| DEATH | DEA|

|%-10.3s| DEATH |DEA |

%% Death %

� Les redirections dans les �chiers s'e�ectuent par les op�erateurs > et >>.

Exemple :

BEGIN {

printf("\t\t\"THE FIELDS OF THE NEPHILIM\"\n") > "newfields"

printf("\t\t -------------------------- \n\n") > "newfields"

}

{printf(" * %-43s *\n", $0) > "newfields"}

END {

printf("\n\t\tLast song of \"The Nephilim\"\n") > "newfields"

printf("\t\t---------------------------\n\n") > "newfields"

}

Resultat dans le fichier "newfields"

"THE FIELDS OF THE NEPHILIM"

--------------------------

* Would you pay lifes pleasure *

* to see me *

* Does it heart for I want you to remain *

* I run your hair through, in another decade *

* Summerlands holds me in sumerian haze *

* Between the spaces, along the wall *

* appearing faces that disappear at dawn *

* Were getting closer, I can see the door *

* closer and closer *

* Kthulhu calls *

* forever remain, forever remain *

Last song of "The Nephilim"

---------------------------

On peut �ecrire dans deux �chiers �a la fois; on peut lire et �ecrire dans deux �chiers
di��erents mais on ne peut �ecrire et lire dans un même �chier.
Exemple : {print $1, ($2 > $3)} Lors de la lecture du �chier d'entr�ee, le premier
champ de chaque enregistrement est �ecrit sur le standard output et le deuxi�eme
champ est �ecrit dans le �chier dont le nom est la valeur du troisi�eme champ.

� Les sorties dans les pipes sont possibles uniquement si le syst�eme supporte les
pipes. L'instruction :

print | "commande"

redirige la sortie de "print" vers l'entr�ee du pipe associ�ee �a "commande ". Les pipes
sont similaires �a ceux employ�es sous Shell. La commande est mise sous quotes.

Exemple :

$1 ~ /(ai|os)/ {print $1 | "tr [a-z] [A-Z]"}

$NF ~ /(ai|os)/ {print $NF | "tr [a-z] [A-Z]"}

REMAIN

CLOSER

CLOSER

REMAIN

� Il est n�ecessaire de fermer les pipes ou les �chiers que l'on a ouvert en lecture
ou en �ecriture si on veut �a nouveau s'en servir dans le même programme ou dans
un programme di��erent. Dans l'exemple pr�ec�edent, on doit fermer le pipe de la
commande par :

close("tr [a-z] [A-Z]")

Si on veut �ecrire dans un �chier apr�es l'avoir lu dans un programme, il est n�ecessaire
de fermer le �chier en lecture par :

close("fichier")

Nous verrons quelques exemples dans la partie Int�eraction avec d'autres programmes.

2.2.4 Les entr�ees

Il y a plusieurs fa�cons d'invoquer des entr�ees vers un programme AWK. La plus
courante est celle que nous avons employ�ee tout au d�ebut du chapitre 1, c.a.d.
utiliser "awk" en tant que commande UNIX ; s'il n'y a pas de �chier en entr�ee, le
standard input est pris par d�efaut.

awk 'programme' fichier_entree ou arguments

awk 'programme' standard input

En tant que commande UNIX, l'entr�ee peut aussi être un pipe.

Exemple : la commande "grep" recupere une ligne du fichier "fields"

et un message s'affiche alors a l'ecran.

grep 'forever' fields | awk '{print "Death will come"

print "To cover all your sad wishes"

$NF="in your faces"

print $0}'

Death will come

To cover all your sad wishes

forever remain, forever in your faces

� Les s�eparateurs sur l'entr�ee :
Il y a deux s�eparateurs d'entr�ee comme il y a deux s�eparateurs de sortie : FS (Field
Separator �equivalent �a OFS) le s�eparateur de champs et RS (Record Separator
�equivalent �a ORS) le s�eparateur d'enregistrements qui en g�en�eral est "\n", aussi les
termes de lignes et d'enregistrement sont synonymes.
FS qui par d�efaut vaut " " peut être modi��e en une expression r�eguli�ere toujours
mise entre quotes. Alors, la substring non nulle la plus �a gauche et la plus longue
match�ee par l'expression r�eguli�ere deviendra le s�eparateur de champ dans la ligne
courante. A chaque fois que l'expression r�eguli�ere matche une substring, NF est
increment�e et la substring devient la chaine nulle "". A chaque enregistrement, NF
est r�einitialis�e �a 1.

Exemple : FS est initialise a "a" ou a la virgule

suivi d'une suite de blancs

BEGIN { FS="(,[ ]*|a)" }

{for (i=1;i<=NF;i++) {printf("%s ", $i)}

printf("%d\n",NF)

}

Would you p y lifes ple sure 3

to see me 1

Does it he rt for I w nt you to rem in 4

I run your h ir through in nother dec de 5

Summerl nds holds me in sumeri n h ze 4

Between the sp ces long the w ll 5

ppe ring f ces th t dis ppe r t d wn 9

Were getting closer I c n see the door 3

closer nd closer 2

Kthulhu c lls 2

forever rem in forever rem in 4

FS peut aussi être modi��e directement dans la ligne de commande en ajoutant
l'option -F. Sur l'exemple pr�ec�edent, on peut donc ex�ecuter le programme contenu
de le �chier "�chierprog" en ayant pr�ealablement supprim�e l'action BEGIN :

Le fichierprog se presente donc maintenant comme :

{for (i=1;i<=NF;i++) {printf("%s ", $i)}

printf("%d\n",NF)

}

Commande de lancement du programme :

awk -F',[ ]*|a' -f fichierprog fields

Remarque : les di��erentes versions de awk ne pr�evoient pas l'utilisation d'une
expression r�eguli�ere apr�es l'option -F; souvent il est seulement possible d'�ecrire un
seul caract�ere entre quotes apr�es l'option -F (Ex : -F'a' place FS="a").

RS d�e�nit la s�eparation entre chaque enregistrement et vaut par d�efaut "\n" �equivalant
donc �a une ligne. Si on veut faire des enregistrements de plusieurs lignes, on peut
l'initialiser �a "" et par exemple d�e�nir la ligne compl�ete comme un champ en prenant
FS="\n".

� La fonction getline :
La fonction "getline " peut être utilis�ee pour lire des enregistrements en entr�ee. Elle
cherche l'enregistrement suivant et remet �a jour les di��erentes valeurs de NR, NF
et FNR d�ependant de l'enregistrement lu et du programme. Elle retourne 1 si un
enregistrement est e�ectivement pr�esent, 0 si la �n de �chier est rencontr�ee ou -1 si
une erreur se produit.
Nous explicitons dans le tableau suivant les possibilit�es d'occurrence de la fonction
getline et leur signi�cation :

EXPRESSION SIGNIFICATION

getline lit sur le standard input et remplit
les variables $0,NR,NF et FNR

getline variable lit l'enregistrement sur le standard input
et le met dans variable tout en remettant
�a jour NR et FNR

getline < "�chier" lit l'enregistrement suivant dans "�chier"
et le met dans $0 tout en remplissant NF

getline variable < "�chier" lit l'enregistrement suivant dans "�chier"
et le met dans variable

commande | getline lit la sortie de commande et le place dans $0
tout en remettant �a jour la valeur de NF

commande | getline variable lit la sortie de commande

et le place dans variable

L'expression getline <"fichier" lit �a partir du �chier "�chier" la ligne courante
plutôt qu'�a partir du standard input. L'expression ne modi�e pas NR et FNR mais
installe la nouvelle valeur de NF.
L'expression getline x<"fichier" lit �a partir du �chier "�chier" le prochain enreg-
istrement et le place dans la variable x . L'expression ne modi�e pas NF, NR et FNR.

Exemple : quand on rencontre l'expression #include "fields" dans un programme,
on decide de remplacer #include "fields" par le contenu du �chier "fields".

Le fichier programme "fichierprog" se presente de la facon suivante :

/^#include/ {

gsub(/"/,"",$2)

while (getline x <$2 >0)

print x

next

}

{ print }

Le fichier "f" se presente de la facon suivante :

#include "fields"

a world without end where no soul can descend

there will be no sumertime how lost lifes been afraid of waking up

so afraid to take the dream shapes of angels the night casts lie dead

but dreaming in my past and their here they want to meet you

they want to play with you so take the dream can't break free

and I hear them call they want to plange you their here once more

they want to take you to the shame of your past

take the dream take me lead me far away take me there

I'll fade away but I can't hide.

On lance la commande awk -f fichierprog f et on obtient :

Would you pay lifes pleasure

to see me

Does it heart for I want you to remain

I run your hair through, in another decade

Summerlands holds me in sumerian haze

Between the spaces, along the wall

appearing faces that disappear at dawn

Were getting closer, I can see the door

closer and closer

Kthulhu calls

forever remain, forever remain

a world without end where no soul can descend

there will be no sumertime how lost lifes been afraid of waking up

so afraid to take the dream shapes of angels the night casts lie dead

but dreaming in my past and their here they want to meet you

they want to play with you so take the dream can't break free

and I hear them call they want to plange you their here once more

they want to take you to the shame of your past

take the dream take me lead me far away take me there

I'll fade away but I can't hide.

L'exemple qui suit ex�ecute la commande UNIX "who" et pipe le r�esultat vers
getline. Chaque it�eration du while lit une ligne de la liste des utilisateurs logg�es et
place la ligne dans la variable d. Puisque NF n'est pas modi��e, pour r�ecup�erer le
nom des utilisateurs, nous sommes oblig�es d'�eclater la ligne contenue dans la variable
d selon le s�eparateur de champ habituel FS. Nous avons alors le nom de l'utilisateur
dans le premier �el�ement du tableau ainsi cr�e�e par l'�eclatement de d. Ce programme
se lance par la commande awk -f fichierprog

BEGIN {while ("who" | getline d)

{

split(d,a)

if (a[1]~/fredo/)

{

gsub(/f/,"F",a[1])

printf("\t%s, Bozo et sa Basse\n", a[1])

}

}

}

Fredo, Bozo et sa Basse

Voyez la di��erence avec le programme suivant o�u nous n'utilisons plus la variable d;
NF est cette fois remis �a jour.

BEGIN {while ("who" | getline)

{

if ($1~/fredo/)

{

gsub(/f/,"F",$1)

printf("\t%s, Caro et ses gros lolos\n", $1)

}

}

}

Fredo, Caro et ses gros lolos

Remarque : dans tous les cas o�u "getline " est utilis�ee, il se peut qu'une erreur se pro-
duise si le �chier �a acc�eder n'existe pas. Ainsi �ecrire while (getline<"fichier") ...

peut être dangereux et peut tourner en rond ind�e�niment si le �chier "�chier"
n'existe pas. Nous conseillons d'utiliser while (getline<"fichier">0) qui teste
si aucune erreur se produit dans la lecture du �chier.

� Les variables dans la ligne de commande :
Comme nous l'avons d�ej�a vu, une ligne de commande AWK peut avoir plusieurs
formes :

awk 'programme' f1 f2 ...

awk -f fichierprog f1 f2 ...

awk -F'separateur' 'programme' f1 f2 ...

awk -F'separateur' -f fichierprog f1 f2 ...

Dans ces lignes de commande, f1, f2 ... repr�esentent des arguments qui sont nor-
malement des �chiers.

� Les arguments dans la ligne de commande :
Les arguments d'une ligne de commande AWK sont accessibles via le tableauARGV.
La valeur de la variable ARGC est �egale au nombre d'arguments plus un.
Ecrivant la ligne de commande :

awk -f fichierprog b bog=roumain d

ARGC a la valeur 4, ARGV[0] contient awk, ARGV[1] contient b, ARGV[2] contient
bog=roumain et ARGV[3] contient d. ARGC est �egal au nombre d'arguments plus
un car le nom de la commande awk est l'argument z�ero.

Exemple :

awk -F'a' '$2=="ppe" {print;

print "ARGC=" ARGC;

for (i=0;i<=ARGC-1;i++)

print "ARGV[" i "]=" ARGV[i]

}' fields

appearing faces that disappear at dawn

ARGC=2

ARGV[0]=awk ---> les options et le programme

ne sont pas prises en compte

ARGV[1]=fields

Les arguments pass�es dans la ligne de commande peuvent être utilis�es en tant que
variable uniquement dans les actions BEGIN, sinon ils sont trait�es comme des noms
de �chier et chaque argument est pass�e dans la variable FILENAME dont le contenu
est mis en entr�ee du programme.

Exemple : On place le programme qui suit dans "�chierprog" et on le lance en pas-
sant deux fois le �chier "�elds" en entre�e. On utilise aussi le �chier "f" pr�ec�edemment
cr�e�e.

BEGIN {

print "Actions BEGIN :"

print "ARGC=" ARGC

for (i=0;i<=ARGC-1;i++) print "ARGV[" i "]=" ARGV[i]

print "FILENAME=" FILENAME

FILENAME="f"

print "FILENAME=" FILENAME

print "\nActions sur les arguments en tant que fichier :"

print "FILENAME reprend la valeur de ARGV[1]"

}

$0~/^a/ {print "FILENAME=" FILENAME;print}

END {

print "\nActions END :"

FILENAME="f"

print "FILENAME=" FILENAME

while (getline<"f">0)

{if ($0~/^a/) print}

}

Actions BEGIN :

ARGC=3

ARGV[0]=awk

ARGV[1]=fields

ARGV[2]=fields

FILENAME=fields

FILENAME=f

Actions sur les arguments en tant que fichier :

FILENAME reprend la valeur de ARGV[1]

FILENAME=fields

appearing faces that disappear at dawn

FILENAME=fields

appearing faces that disappear at dawn

Actions END :

FILENAME=f

a world without end where no soul can descend

and I hear them call they want to plange you their here once more

2.2.5 Les Int�eractions avec les autres programmes

La fonction AWK system (expression) ex�ecute la valeur de la chaine expression

prise comme commande. La valeur retourn�ee est celle de la commande ex�ecut�ee.

Exemple : On prend les deux �chiers suivants :

FICHIER1 : FICHIER2 :

Je suis le fichier 1 Je suis le fichier 2

Le programme qui suit int�eragit avec la commande UNIX "cat"

BEGIN {

while (1)

{

printf("\tFichier a lire \n")

printf("\tQuitter : Q\n")

Reponse=convert()

print Reponse

if (Reponse!="Q")

{

system("cat " Reponse)

}

else exit

}

}

function convert ()

{

getline chaine

gsub(/'/,"\\'",chaine)

"echo "chaine" | tr '[a-z]*' '[A-Z]*'" | getline CHAINE

return (CHAINE)

}

La fonction "convert" convertit tous les caract�eres en majuscules.

Ex�ecution :

awk -f fichierprog

Fichier a lire

Quitter : Q

fichier1

FICHIER1

Je suis le fichier 1

Fichier a lire

Quitter : Q

fichier2

FICHIER2

Je suis le fichier 2

Fichier a lire

Quitter : Q

fichier1

FICHIER2

Je suis le fichier 2

Attention, nous avons oubli�e de fermer le pipe de
"echo "chaine" | tr '[a-z]*' '[A-Z]*'"

Il est necessaire de le fermer par l'instruction :
close("echo "chaine" | tr '[a-z]*' '[A-Z]*'")

La fonction convert devient :

function convert () {

getline chaine

gsub(/'/,"\\'",chaine)

"echo "chaine" | tr '[a-z]*' '[A-Z]*'" | getline CHAINE

close("echo "chaine" | tr '[a-z]*' '[A-Z]*'")

return (CHAINE)

}

Nous pouvons r�ealiser une commande Shell �a partir d'un programme AWK. Dans
la plupart de nos exemples, nous avons utilis�e l'option -f ou nous avons plac�e le pro-
gramme �a ex�ecuter entre quotes : awk '{print $1}' ...

Pour r�eduire le nombre d'instruction �a taper, nous pouvons mettre la ligne de com-
mande et le programme dans un �chier ex�ecutable que l'on invoquera en tapant
uniquement le nom du �chier.

awk '{print $1}' $* est place dans le fichier "champ1"

chmod +x champ1 pour rendre "champ1" executable

champ1 liste de fichiers execute le programme sur

la liste des fichiers

Supposons maintenant que l'on veuille cr�eer une commande champ qui �ecrit une
combinaison arbitraire de champs sur chacun des �chiers en entr�ees :

champ 2 5 7 ... fichier1 fichier2 ...

qui ecrirait les 2eme, 5eme et 7eme champs de

chaque enregistrement, pour chaque fichier en

entree.

Comment reconnaitre les num�eros de champ des �chiers ? On peut le faire grâce �a
la variable ARGV :

awk '

BEGIN {

for (i=1;ARGV[i]~/^[0-9]+$/;i++)

{

champ[++nf]=ARGV[i]

ARGV[i]=""

}

if (i>=ARGC) # pas de noms de fichier

ARGV[ARGC++]= "-" # donc force stdin .

}

{ for (i=1;i<=nf;i++)

printf("%s%s", $champ[i], i < nf ? " " : "\n")

}

' $*

champ 2 4 6 fields f

you lifes

see

it for want

run hair in

holds in haze

the along wall

faces disappear dawn

getting I see

and

calls

remain, remain

world end no

will no how

afraid take dream

dreaming my and

want play you

I them they

want take to

the take lead

fade but can't



86 CHAPTER 2. LE LANGAGE AWK EN D�ETAIL



Chapter 3

Applications

87



88 CHAPTER 3. APPLICATIONS



89

Ce chapitre contient quelques exemples d'applications en AWK et comment on
peut utiliser le langage AWK. Les programmes sont des r�esultats d'�etudes prolong�ees
men�ees sur la nature qui nous environne.
AWK peut jouer un rôle non n�egligeable dans plusieurs domaines tels que :

Validation, transformation, traitement de donn�ees
Gestion de base de donn�ees
G�en�eration de texte, mise �a des formats sp�eciaux
Compilation, Interpr�etation, Transformation de langages
Exp�erimentation sur des algorithmes
........



90 CHAPTER 3. APPLICATIONS

3.1 Base de donn�ees

3.1.1 Introduction

Cette partie montre comment AWK peut être utilis�e pour extraire des renseigne-
ments et g�en�erer des r�esultats �a partir de donn�ees stock�ees dans des �chiers.
Une certaine structure de donn�ees a �et�e choisie dans notre exemple mais il est clair
que la technique peut être appliqu�ee �a n'importe quelle structure �a partir du moment
o�u elle est bien sp�eci��ee et o�u elle ne varie pas al�eatoirement.

3.1.2 Sp�eci�cations

Nous avons choisi un exemple r�eel d'interrogation d'une base de donn�ees qui pourrait
être pr�esente dans n'importe quelle discoth�eque. Elle regroupe par style de musique,
les groupes, les titres des albums, le nombre de disques selon le type (cassette, CD
ou disque vinyl) pr�esent dans le magasin et les prix associ�es.
Le �chier se pr�esentera alors de la fa�con suivante :

STYLE : COLD

GROUPE : THE FIELDS OF THE NEPHILIM

ALBUM : THE NEPHILIM: ELIZIUM: DAWNRAZOR

SUPPORT: CD=568; K7=54: VINYL=21: K7=1365

PRIX : 87; 67: 34: 72

...

On classe dans la section SUPPORT la disponibilit�e pr�esente en magasin par type :
cassette, CD, disque vinyl et lorsqu'il n'y a plus de disque dans un certain type, le
nombre et le prix seront ignor�es :

Exemple :

Dans le cas precedent, la disponibilite pour chaque album est de :

THE NEPHILIM : CD ----> 568 au prix de 87 F par unite

K7 ----> 54 au prix de 67 F par unite

ELIZIUM : VINYL ----> 21 au prix de 34 F par unite

DAWNRAZOR : K7 ----> 1365 au prix de 72 F par unite

On e�ectue une interrogation par style de musique (liste de tous les groupes avec
les renseignements associ�es dans le style de musique demand�e), par groupe (liste de
tous les albums pr�esents dans la base, le nombre disponible et leur prix) et par titre
d'album (prix, disponibilit�e, groupe et style) :

---> Style de musique

---> Groupe

---> Titre d'album

---> Quitter

Tant que le choix Quitter n'est pas choisie, le programme bouclera et chaque
r�esultat d'interrogation sera �a la fois a�ch�e sur la sortie standard et mis dans un
�chier "out.data" o�u tous les r�esultats seront m�emoris�es.
Nous permettons aussi la modi�cation de la base de donn�ees.

3.1.3 Le �chier de donn�ees

Le �chier de donn�ees est denomm�e musique. Nous avons choisi une structure de
donn�ees avec les particularit�es suivantes :

- Tous les caract�eres doivent être en majuscule.
- Le champ s�eparateur entre les �el�ements virtuels est ": "
qui est initialis�e en d�ebut de programme.
- Le champ de d�etection du style de musique est "�".
- Le champ "; " est un sous-champ de d�etection �a l'int�erieur
d'un �el�ement virtuel qui permet de distinguer le type
de support et le prix qui lui est associ�e.

Il se pr�esente comme suit :

*

STYLE : COLD

GROUPE : THE CURE

ALBUM : BOYS DON'T CRY: SEVENTEEN SECONDS: PORNOGRAPHY: FAITH

SUPPORT: CD=23; VINYL=485; K7=354: VINYL=29; K7=45: K7=384: CD=543; K7=453

PRIX : 102; 46; 87: 34; 74: 70: 103; 75

GROUPE : THE FIELDS OF THE NEPHILIM

ALBUM : THE NEPHILIM: ELIZIUM: DAWNRAZOR

SUPPORT: CD=568; K7=54: VINYL=21: K7=1365

PRIX : 87; 67: 34: 72

GROUPE : JOY DIVISION

ALBUM : STILL: TO THE CENTER

SUPPORT: VINYL=365: CD=1356; VINYL=132; K7=827

PRIX : 43: 100; 65; 43

GROUPE : BAUHAUS

ALBUM : BELLALUGO IS DEAD

SUPPORT: CD=5439; K7=324

PRIX : 74; 62

GROUPE : THE SISTERS OF MERCY

ALBUM : MARIAN: TEMPLE OF LOVE

SUPPORT: CD=3565; VINYL=2356; K7=543: CD=453; VINYL=2341; K7=3545

PRIX : 86; 41; 72: 78; 43; 69

GROUPE : THIS MORTAIL COIL

ALBUM : DEATH SHALL COME

SUPPORT: CD=654

PRIX : 83

GROUPE : DEAD CAN DANCE

ALBUM : THIS NIGHT

SUPPORT: CD=654; K7=5643

PRIX : 92; 74

GROUPE : THE MISSION

ALBUM : CARVENDISH

SUPPORT: CD=743; VINYL=764

PRIX : 84; 34

*

STYLE : HARD

GROUPE : GUN'S N ROSES

ALBUM : WELCOME TO THE JUNGLE: DON'T CRY

SUPPORT: CD=653; VINYL=765: CD=931; VINYL=126; K7=43211

PRIX : 93; 31: 85; 29; 58

GROUPE : WHITE LION

ALBUM : I DREAM OF YOU

SUPPORT: VINYL=439; K7=5742

PRIX : 42; 65

GROUPE : AC/DC

ALBUM : TOUCH TOO MUCH: FOR THOSE ABOUT TO ROCK: BACK IN BLACK

SUPPORT: CD=568; K7=54: VINYL=21: K7=1365

PRIX : 87; 67: 34: 72

GROUPE : TRUST

ALBUM : MARCHE OU CREVE: MISSIONNAIRE: BON SCOTT

SUPPORT: CD=578; K7=84: VINYL=51: K7=7365

PRIX : 87; 62: 33: 76

GROUPE : TELEPHONE

ALBUM : LA BOMBE HUMAINE: ARGENT TROP CHER

SUPPORT: CD=541; VINYL=877: CD=2346; VINYL=2346; K7=133

PRIX : 93; 31: 85; 29; 58

GROUPE : SCORPIONS

ALBUM : I'M STILL LOVING YOU

SUPPORT: VINYL=432; K7=371

PRIX : 42; 65

*

STYLE : TECHNO

GROUPE : FRONT 242

ALBUM : CRAZY

SUPPORT: VINYL=432; K7=371

PRIX : 42; 65

GROUPE : ALIEN SEX FIEND

ALBUM : SEX IS BEAUTIFUL

SUPPORT: CD=5439; K7=324

PRIX : 74; 62

*

STYLE : ROCK

GROUPE : NEW MODEL ARMY

ALBUM : IMPURITY: I'M PRIED OF YOU

SUPPORT: CD=3545; K7=466: VINYL=768; K7=254

PRIX : 86; 63: 31; 43

GROUPE : ALAN PARSON PROJECT

ALBUM : EYE IN THE SKY: STEREOTOMY

SUPPORT: CD=3545; K7=466: VINYL=768; K7=254

PRIX : 86; 63: 31; 43

GROUPE : PINK FLOYD

ALBUM : THE WALL: THE DARK SIDE OF THE MOON: OBSCURED BY CLOUDS: ANIMALS: RELICS

SUPPORT: CD=543; K7=543: VINYL=54; K7=320: CD=346; VINYL=3545: K7=32: CD=3456; K7

PRIX : 91; 62: 31; 68: 79; 39: 67: 68; 58: 81: 37; 52: 74: 41; 58: 62

GROUPE : BARCLAY JAMES HARVEST

ALBUM : VICTIMS OF CIRCUMSTANCE

SUPPORT: CD=5439; K7=324

PRIX : 74; 62

*

3.1.4 Programme

Ce programme fonctionne exclusivement sur un syst�eme tel que UNIX et montre
aussi ses limites. Le programme fnac est le suivant :

BEGIN {FS="\: "

while ((Reponse !~ /^( )*Q( )*$/) && (decision !~ /^( )*Q( )*$/))

{

printf("\n\tRecherche ou Mise a jour -----> R , M\n")

printf("\t QUITTER -----> Q\n")

getline decision

if (decision ~ /^( )*R( )*$/)

{

system("rm out.data")

Reponse=""

while ((Reponse !~ /^( )*Q( )*$/) && (Reponse !~ /^( )*P( )*$/))

{

printf("\n\n")

printf("\tRecherche par groupe -----> 1\n")

printf("\t par album -----> 2\n")

printf("\t par style de musique -----> 3\n\n")

printf("\tRevenir a l'etat precedent -----> P\n\n")

printf("\t QUITTER -----> Q\n")

getline Reponse

if (Reponse ~ /^( )*1( )*$/) groupe()

if (Reponse ~ /^( )*3( )*$/) style()

if (Reponse ~ /^( )*2( )*$/) album()

}

close("out.data")

print "\tVotre choix se trouvent aussi dans le fichier \"out.data\"\n"

}

else if (decision ~ /^( )*M( )*$/)

{

print "\tVous allez modifier sous l'editeur \"vi\" d'UNIX la base"

print "\tMais ATTENTION a respecter la syntaxe."

print "\t Maintenant taper V \n"

getline vi

if (vi ~ /^( )*V( )*$/) system("vi musique")

}

}

}

function groupe()

{

printf("Nom du groupe?\t")

G=convert()

trouve=0

while (getline<"musique">0 && trouve==0)

if ($1 ~ /GROUPE/ && $2==G)

{

trouve=1

affichage()

}

if (trouve==0)

printf("\n\tLe groupe ne se trouve pas dans la base\n\n")

close("musique")

return(trouve)

}

function style()

{

printf("Style de musique?\t")

ST=convert()

trouve=0

while (getline<"musique">0 && trouve==0)

if ($1 ~ /STYLE/ && $2==ST)

{

trouve=1

getline<"musique"

while ($1!="*")

{

getline<"musique"

affichage()

}

}

if (trouve==0)

printf("\n\tCe style de musique n'existe pas\n\n")

close("musique")

return

}

function album()

{

printf("Titre de l'album recherche?\t")

A=convert()

trouve=0

while (getline<"musique">0 && trouve==0)

{

if ($1 ~ /STYLE/) {sty=$2}

if ($1 ~ /GROUPE/) {gr=$2}

if ($1 ~ /ALBUM/)

{

i=2

while (i<=NF && trouve==0)

if ($i==A)

trouve=1

else i++

}

}

if (trouve==0) printf("\n\tL'album n'est pas dans la base\n\n")

else

{

ind_cd = ind_v =ind_k7 = 0

decode_sup()

decode_prix()

printf("\n\tAlbum recherche : %s\n",A) >>"o.data"

printf("\n\tNom du groupe : %s\n",gr) >>"o.data"

printf("\tStyle de musique : %s\n\n",sty) >>"o.data"

printf("\tLe nombre de CD disponibles : %d au prix : %d FF\n",nb_cd\

,prix[ind_cd]) >>"o.data"

printf("\tLe nombre de VINYL disponibles : %d au prix : %d FF\n",\

nb_v,prix[ind_v]) >>"o.data"

printf("\tLe nombre de K7 disponibles : %d au prix : %d FF\n",nb_k7\

,prix[ind_k7]) >>"o.data"

print "********************************************************"\

>>"o.data"

close("o.data")

system("cat o.data")

system("cat o.data >> out.data")

system("rm o.data")

}

close("musique")

return

}

function affichage ()

{

if ($1 ~ /GROUPE/)

{

printf("\n\tNom du groupe : %s\n",$2) >>"o.data"

getline<"musique"

printf("\tTitre des albums : \n") >>"o.data"

for (i=2;i<=NF;i++)

printf("\t\t %s\n",$i) >>"o.data"

getline <"musique"

print "*********************************************************" >>\

"o.data"

close("o.data")

system("cat o.data")

system("cat o.data >> out.data")

system("rm o.data")

print "\n"

}

return

}

function convert()

{

getline chaine

gsub(/'/,"\\'",chaine)

"echo "chaine" | tr '[a-z]*' '[A-Z]*'" | getline C

close("echo "chaine" | tr '[a-z]*' '[A-Z]*'")

return(C)

}

function decode_sup()

{

nb_cd = nb_v = nb_k7 = 0

split($i,sup,"; ")

for (j in sup)

{

split(sup[j],nb,"=")

if (nb[1] ~ /CD/) { nb_cd = nb[2] ; ind_cd = j }

if (nb[1] ~ /VINYL/) { nb_v = nb[2] ; ind_v = j }

if (nb[1] ~ /K7/) { nb_k7 = nb[2] ; ind_k7 = j }

}

return

}

function decode_prix()

{

getline <"musique"

split($i,prix,"; ")

return(prix)

}

Le programme utilise la plupart des pouvoirs du langage AWK sur la gestion
d'un �chier de donn�ees. On y rencontre explicitement l'utilisation :

1. des expressions r�eguli�eres par exemple dans l'acquisition des r�eponses donn�ees
par l'utilisateur sous toute leur forme : majuscule ou minuscule, s�epar�ees par
des blancs.

2 . des fonctions pr�e-d�e�nies AWK par exemple dans la construction des sous-
champs avec la fonction split.

3 . des fonctions cr�e�ees par l'utilisateur par exemple la fonction convert () qui con-
vertit toute chaine de caract�eres en majuscule tout en supprimant les blancs
et qui renvoie la chaine ainsi modi��ee.

4 . la majorit�e des entr�ees-sorties en AWK : la fonction getline lisant le standard
input (acquisition de donn�ees de l'utilisateur) ou le �chier musique (acc�es �a
la base de donn�ees), les pipes, les fonctions print et printf redirig�ees vers un
�chier (o.data ou out.data) ou vers le standard output et en y associant la
fonction close ().

5 . des fonctions UNIX par la fonction system () ou directement en entr�ee de la
fonction getline pour �eviter de construire des fonctions existant d�ej�a dans le
syst�eme UNIX.

Nous donnons un exemple d'ex�ecution du programme �a la page suivante.
Nous lan�cons le programme par awk -f fnac

Recherche ou Mise a jour -----> R , M

QUITTER -----> Q

R

Recherche par groupe -----> 1

par album -----> 2

par style de musique -----> 3

Revenir a l'etat precedent -----> P

QUITTER -----> Q

1

Nom du groupe? The Cure

Nom du groupe : THE CURE

Titre des albums :

BOYS DON'T CRY

SEVENTEEN SECONDS

PORNOGRAPHY

FAITH

*********************************************************

Recherche par groupe -----> 1

par album -----> 2

par style de musique -----> 3

Revenir a l'etat precedent -----> P

QUITTER -----> Q

2

Titre de l'album recherche? Don't cry

Album recherche : DON'T CRY

Nom du groupe : GUN'S N ROSES

Style de musique : HARD

Le nombre de CD disponibles : 931 au prix : 85 FF

Le nombre de VINYL disponibles : 126 au prix : 29 FF

Le nombre de K7 disponibles : 43211 au prix : 58 FF

********************************************************

Recherche par groupe -----> 1

par album -----> 2

par style de musique -----> 3

Revenir a l'etat precedent -----> P

QUITTER -----> Q

P

Votre choix se trouvent aussi dans le fichier "out.data"

Recherche ou Mise a jour -----> R , M

QUITTER -----> Q

M

Vous allez modifier sous l'editeur "vi" d'UNIX la base

Mais ATTENTION a respecter la syntaxe.

Maintenant taper V

V

-------> MODIFICATIONS.........

3.2 Des mini-langages

3.2.1 Introduction

AWK est souvent utilis�e pour d�evelopper des interpr�eteurs, des assembleurs et autres
exp�erimentations sur la syntaxe ou la s�emantique de mini-langages, c.a.d. des lan-
gages destin�es �a des applications sp�eci�ques.
Nous avons choisi dans cet exemple de g�en�erer un calculateur.

3.2.2 Sp�eci�cations

Les expressions arithm�etiques avec les op�erateurs +,-,� et / peuvent être d�ecrites
par la grammaire :

expr ----> terme

expr + terme

expr - terme

terme ----> fact

terme * fact

terme / fact

fact ----> nombre

( expr )

Nous pouvons alors �ecrire des expressions du type 1 + 2 � 3 qui seront analys�ees de
la fa�con suivante :

1 -- nombre -- fact -- terme -- expr --------\

+ ------------------------------------------- expr

2 -- nombre -- fact -- terme ---\ /

* ------------------------------ terme -----/

3 -- nombre -- fact ------------/

Pour r�ealiser le calculateur, nous avons besoin d'un "Parser" pour les expressions.
On �ecrit une fonction pour chaque �el�ement non terminal de la grammaire expr pour
expr, terme pour terme et fact pour fact.

3.2.3 Programme

BEGIN {

print "\tProgramme de la calculatrice"

print "\tLaissez un blanc entre les operateurs et les operandes.\n"

print "\tPour QUITTER ------------------------> CTRL/C\n"

}

NF > 0 {f = 1

e = expr()

if (f <= NF) printf("erreur en %s\n",$f)

else printf("\t%.8g\n",e)}

function expr( e) # terme | terme [+-] terme

{e = terme()

while ($f == "+" || $f == "-")

e = $(f++) == "+" ? e + terme() : e - terme()

return e}

function terme( e) # fact | fact [*/^] fact

{e = fact()

while ($f == "*" || $f == "/" || $f == "^")

if ($(f++) == "*") e = e * fact()

else if ($(f-1) == "/") e = e / fact()

else e = e ^ fact()

return e}

function fact( e) # chiffre | (expr)

{if ($f ~ /^[+-]?([0-9]+[.]?[0-9]*|[.][0-9]+)$/)

return $(f++)

else if ($f == "(")

{

f++

e = expr()

if ($(f++) != ")")

print("erreur : on atendt ) en %s\n",$f)

return e

}

else

{

printf("erreur : on attend un numero ou ( en %s\n",$f)

return 0

}}

